
Introduction

This cookbook represents both the knowledge Geert De Peuter has obtained as a customer using PATROL and as a Developer at BMC Software and from the thousands of questions that have been asked at the BMC Software Developer Connection forums.  

The cookbook has been written in a question answer format.  We recommend its use as a reference to aid you in your development efforts.

To keep up to date on the latest questions and answers join us at http://devcon.bmc.com and select the user forums.

The PATROL Virtual Machine

Before talking about anything else, it is very important to understand how a PATROL Agent really behaves on the system on which it is installed.

What is a Virtual Machine?

"A self-contained operating environment that behaves as if it is a separate computer."

This means that the PATROL Agent is really an OS running on the host OS., and unlike some VM's, the PATROL Virtual machine has access to the host operating system. 

An enormous advantage of this design is that you only need to know one OS if you want to write KMs.  If you understand how the VM works, you know how the agent will behave on any host operating system.  Since the VM is a self-contained operating environment, it also has its own native programming language (as we all know : PSL).

It is possible to use any other programming language for programming in the VM, as long as this language is somehow available on the host system.  It is also possible for the VM to interact with existing command line processors for applications.

Because of that, it is fairly easy from within PATROL to talk to the host OS (sh, cmd,…), other applications running on the host system (sql) or even other languages that have been installed on the host system (perl, rexx, scheme, python, java…) .

Picture of the PATROL Agent VM :


The code the PSL compiler generates is the micro code for the PATROL VM (quad code).  The optimizer can optimize this code.

From then on, you only have Virtual Machine instructions left in your code.  The interpreter is the same as the instruction processor on your system.  In this case, we might call it a Virtual instruction Processor will send the code to an execution engine.  The execution engine will make sure the instruction executes (this does not necessarily mean we are executing code on the host OS.  You will have seen that only a few calls in PSL will actually do something on the host OS (things like popen, fopen, snmp_...).

A command like (a=+5;) will be executed inside the VM.

Program Lifecycle :



Variables Inside the Virtual Machine

The reason why we explain about the different type of variable you would find inside the VM is because the terminology behind this seems to change from person to person.  For the rest in the document we will not keep explaining what we mean with a configuration variable or a local variable, so here is a summary of all types of variable you would find inside the VM.

Global variables

Global variables are variables that are accessible (known) within the PATROL VM.  They can be accessed from within every PSL process.  They are typically addressed with their PATH definition in the PATROL Agents Object Store (aka heap).

Examples of global variables :

/ipAddress

/MYAPP/myvar

/AgentSetup/defaultAccount

/ORACLE/instances

Different types of global variables

Non permanent

These variables are lost after an agent restart.  They will remain resident in the agent’s memory as long as the agent keeps running.

As a KM developer, you can freely assign any name for a global variable using the get() and set() function.  The unset() function will remove the variable from the agents memory.

These variables are read-write.

The technique of using non-permanent global variables allows developers to pass information from one PSL process in another.

This is used a lot when you want someone to enable debug on an instance.  In that case you can set a variable 

set("/<MYAPP>/<myinst>/parameterdebug","TRUE");

and every parameter can check if debug was turned on 

get("../parameterdebug");

Configuration variables (permanent)

These variables are stored inside the configuration database of the agent.  They will remain resident during the life of the agent.  Even after an agent restart, because the agent will reload it's configuration database when it is started.

You normally access these variables using pconfig() calls.  These calls will always go to the configuration database, and this will cost a lot more than operations on non-permanent variables.

Even a read operation will go to the configuration database.  This is so because the configuration database can be updated from outside the PATROL Agent (using the OS pconfig command).

These variables are read-write.

Note: you might have noticed that a normal get() operation will work on these kind of variables also.  This will have the same cost as using the pconfig() call.

System variables

System variables are variables that are executed on the host OS on demand.  An example of one of these variables is "/ipAddress".

These variables are read-only of course. 

Attributes

An attribute is assigned by PATROL on a certain object.  We can have a discussion if an attribute can be called a global variable.  Let's just consider an attribute to be a very special global variable that PATROL will create for you.  Depending on the type of attribute, these can be read-only or read-write.

Examples of attributes are:

/<MYAPP>/instances

/<MYAPP>/<myinst>/status

A very special attribute of a parameter is the value.  When you assign a new value to this attribute, the agent will store this new value in the history database, if the parameter type is not a text parameter.

The PATROL Agent doesn't store historical data on a text parameter.  This special feature allows you to display a reasonable amount of textual information in a parameter, without the worry that this will cause an overload on the history database.

What people sometimes forget it that this information is actually stored in memory, and you should therefore make sure you are not creating cumulative parameters !

It sounds tempting to create a parameter that displays all the error messages in a log file.  Somewhere in the PSL script you might do something like this

set("value",get("value").new_data);

The consequence is that if the agent is not rebooted regularly (very likely) and your application has a problem, and will write a lot a errors in the log file, the PATROL Agent will keep taking more and more memory.

In this case, it might be better to only display x number of lines, or to implement this feature as a menu command.

Local variables

Local variables are private for each PSL process.  They are initialized at the start of the process.

named

author=“John Doe”;
firstname=nthargf(author,1);

The PATROL Agent will create a variable author an assign it the value "John Doe".  He will then execute the function nthargf(author,1) and assign the result to a second variable named firstname.

Result,  two named local variables are created.

unnamed

firstline=nthline(cat(myfile),1);


Because no name was given for the subfunction "cat(errorfile)", PATROL will assign it a name.  We call these variables unnamed because you don't know the name.

The former example will expand to :

unnamed_var=cat(myfile);
firstline=nthline(unnamed_var,1);


This means that the output of every command is stored inside a variable (named or unnamed).

About Channels

Another piece of theory that is usually not very well understood is about the use of channels. I will try to describe why they are there and how they work.

Private channels

Another way to get data from your application (and using sub-process execution) is to use the connection oriented method.  This allows you to talk to the executable in a conversational way.

The good thing is that the process is spawned only once, when the connection (conversation) is established.

We call the connection a channel.  Normally a channel is private for a PSL process.

The PSL function to establish a channel to a process is popen(CMDTYPE,"cmd");

When the channel is established, you can use functions to talk to the executable through the channel (write()) or listen to your executable (read()).

If you want to finish the communication (and stop the spawned child process) you can close() the channel.

This possiblility can greatly improve the performance of the KM you are writing, especially when the startup of the child process can take a lot of CPU / time.

An example for this would be Oracle’s SQLPLUS.  This is a conversational command.  Instead of writing select statements on stdin, you write it in the channel.  Instead of reading from stdout, you read from the channel.

Shared channels

Once you have created a channel, and you want to use it in multiple PSL processes, the process that created the channel will have to share it.

Sharing a channel means, giving it a name, and placing that name in the "PATROL Shared Memory" space from the Virtual Machine (instead of the process memory area).

This is done with the share() command, that does all of this for you.

When using shared channels, there is only one sub-process that will answer all requests from a number of PSL processes.  This can be very good for performance (applied to a KM that works on Oracle this can mean one connection per Oracle instance, instead of one connection per parameter in the KM).

Serializing shared channels

Since multiple processes can talk at the same time over a shared channel, it can be desirable to introduce channel locking in your KM design.

Sometimes it is not harmful if a command is messed up in the channel, and mostly you can detect that this happened when you read the output of the channel.  In some cases it is not allowed that this happens.  If this is the case, you will have to put locking on a channel.

When you lock a channel, it will be locked for a certain PSL process.  When you implement locking, you normally lock the channel before writing to it, and you unlock it after you read the output from the channel.

If another process wants to write at the channel, he will be waiting until he can acquire a lock on the channel (this will be when the process occupying the channel unlocks it).  If no process is using the channel at the point of request, he will immediately get the lock.

You will also have to implement timeouts on the lock, so a process will not be able to get the channel for an infinite time.  This would block the channel, and your KM will stop responding.

Serialized shared channels and menu commands

You need to be careful when using shared channels in your menu commands.  Normally the user is used to getting immediate feedback when he executes a menu command.  

If the execution of a menu command goes over the same channel than your monitoring, a user issuing a lot of menu commands will influence the monitoring part of your KM (if the user gets the lock all the time).  Another possibility is that the user might not get the lock at all (or within an acceptable time interval).

The users perception will be that your KM performs very badly.

Therefore, it might be advisable to execute your menu commands as one-shot commands, or over another channel, special for menu commands.

Dealing with alarm ranges and recovery actions

KM recovery actions are know to most KM developers use to heal an unhealthy managed object, create reports based on alarm conditions or execute arbitrary code when an object goes into alarm.

This feature allows a PATROL developer to do marvelous things.  The benefits for these actions can be tremendous.  The application administrators have to worry less about the health of their systems.  
Problems are detected more accurately because of the detailed reports. There is an audit trail of what happened on the system.  Decreasing downtime and early problem detection result in higher service levels.

A recovery action is an action that can be executed when the value attribute of a standard or consumer parameter falls within a certain range.  This range can be set using either the parameter properties (GUI) or using the set_alarm_ranges() PSL call.

When defining recovery actions, it is very important to have a good idea of what alarm ranges mean and how the agent works with them.

Alarm ranges

You will all know that you have 3 of them, and they all take two arguments.  It is very important to understand that the arguments (minimum & maximum) are both integers.

· Border
( ] minborder , maxborder [

· Alarm 1
( [ minalarm1,max alarm1 ]

· Alarm 2
( [ min alarm2,max alarm2 ]

With the following restrictions : minborder (  minalarm1 ( max alarm1 ( min alarm2 ( max alarm2 (  maxborder
A graphical representation of this is :


This picture will explain when a value is evaluated to belong to a certain range.  You can see for example that the border is exclusive and if the definition would be ]0,100[ , this means that 0 is not part of the border range, but 100.1 is !

OK, once we know all this, we still have to examine the situation where max alarm1 = min alarm2 .

Well, in that case, Alarm1 will take the overlap value. Sounds to technical, well here is the graph of that situation:


So what happens in the extreme case where all values are the same ?


Recovery actions

Unlike what a lot of people think, a recovery action is defined on a range and not on a certain state.  The unlucky choice of the words “alarm range” made a lot of people think that a recovery action has to be bound to a certain alarm state, and can therefore only be executed on a WARN or ALARM state.

This is not true! A recovery action definition is actually done on a range and the state is merely an attribute of this range.  This means you can define recovery actions on OK states as well if you want to!

When a new value is set, the agent will perform a range check (this explains why you don’t see something going into alarm immediately after you modified the range, the agent will first have to collect a new value).

No matter what happens, there will only be one recovery action running per parameter!  No matter how often you switch between ranges.

If there is a recovery action left to process (and no recovery action is running), the recovery action will fire off immediately.  If there was a recovery action running (even on another range), your recovery action will not execute.

Each range can have multiple recovery actions.  The moment the agent changes range (NOT the same as a state change), he will restart processing the list of recovery actions from the top.

If the range is still the same at set-time, the agent will try to fire of the next recovery action in the list (as long as there is not another recovery action running for that parameter).

This is a logical flow of recovery action execution


Probably it’s better to show you a practical graph explaining when a recovery action is fired.

· Alarm 1 has two recovery actions : ALM1_A lasts 7 seconds, ALM1_B  2 seconds

· Alarm 2 has two recovery actions : ALM2_X lasts 2 seconds, ALM2_Y 2 seconds

· Poll interval 5 seconds

The colored bar indicates the period it takes for the recovery action to execute.


(Micro-) Optimizing your code

Usually, something that keeps most developers busy is trying to optimize code.  It does make one feel more proud of his work after having been able to improve performance.  However, sometimes the optimization turns into an obsession and more and more into micro optimization where you are not really achieving a lot more by making changes to the code that seems to work fine.

Instead of just writing about optimization in general, I thought it would be nicer to see how optimization can happen in practice. 

The problem that we are trying to solve is very simple : ”Find the fastest way to get the last word on a line.  We are just interested in the last word.”.


If you want to look for the last word in a line and you are not interested in the number of words you would type something like this in awk on UNIX

echo $LINE | awk '{print $NF}’

Although this solution sounds simple, it would require multiple processes in UNIX and it is not really platform independent.  Anyway, since this is a book about PATROL, you can imagine that we will be looking for the best possible solution in PSL.

In PSL, straight forward coding would produce this :

# Don't know how much words in a line, but need last word

number_words = 0;

foreach word w (myline) 

{

number_words++;

}

# Now get the word

lastword = ntharg (myline, number_words);

First Optimization

When we look closely at this little PSL program, we can easily see that there is a lot of useless code in the program.  The reason why we are counting the number of words is because we want to use ntharg to get the last work from the line.

If we look carefully, we see that all that counting is not really necessary, because what do you think the value of “w” will be at the end of the program ?

Exactly : The last word !

So our first example can be rewritten as :

foreach word lastword (myline) {}

This will result in an empty loop.

Each time this (empty) loop executes the variable lastword will be assigned the value of the next word in myline.  At the end of the statement, the foreach will exit, but lastword will still have the value of the last word of the line

Prevent looping

We still have a loop, and we know performance might be even better without loops.  So how can we rewrite the code so we don’t have looping anymore

The original code was…

# Don't know how much words in a line, but need last word

number_words = 0;

foreach word w (myline) 

{

  number_words++;

}

# Now get the word

lastword = ntharg (myline, number_words);

In our first optimization step we kept the loop, but just removed everything we didn’t need.  What we will try to do now, is rewrite the code, but without loops.  This means : find another way of getting the number of words without a loop.

We came up with the following solution :

wordlist = ntharg(myline, "1-");

num_words = lines(wordlist);

lastword = ntharg(myline, num_words);

If we compare this with our first optimized version, we replaced one single line of code with three lines of code… but without any loops.  

· The fist statement will break the line in a list of words.

· The second statement will count the number of words.

· The third statement will get the actual word.

If we profile this code, we see that it works faster than the original, but performance is practically the same as our first optimization.

Micro Optimization

We have already optimized the basic code, and probably this is where we should stop, but the purpose of this discussion is to go as deep as we possibly could.

It was clear that our foreach loop wouldn’t bring us further optimization.  Since we know that looping is quite expensive (although we only used one PSL statement in our solution and that was foreach)

The second example (as fast as the empty foreach()) however had three statements, maybe we can optimize something there.  Let’s take a look at our code again.

wordlist = ntharg(myline, "1-");

num_words = lines(wordlist);

lastword = ntharg(myline, num_words);

And yes.  Line two and three can be combined in a single PSL function : tail().  Therefore we found our real performance winner 

wordlist = ntharg(myline, "1-");

lastword = tail (wordlist,1);

When I talk about performance gain between optimized 1,2 or 3, it is more in 1/10 seconds than in real seconds, but the difference between the original and 3 will speed processing with 20-50% of the original time.

Another problem

Now let’s try a variation of the problem :”Find the last word in a text block (that means a block with newlines in it)”

It should be obvious that even our best-optimized example will not really be good enough for this problem.

wordlist = ntharg(myline, "1-");

lastword = tail (wordlist,1);

On a full text block, this piece of code will produce the correct result, but not with the expected performance, because now our complete multi-line text is changed into a list.

Of course a simple modification will solve the problem

last_line = tail(mytext,1);

last_word_list = ntharg(last_line, "1-")

last_word = tail (last_word_list,1);

Now if you want to offer this as a function in a library that more than just you will use, you could ask yourself which of the optimized versions you should use.

The answer is very simple: make sure to implement the last example, that will perform best for the worst case scenario and reasonably good for the best case scenario.

Indeed, the end user of your library doesn’t know how your function will behave with a certain input (and he shouldn’t care).  The enduser will expect a reasonable result, no matter what the input will be.

Just as a comparison :

· In cases where you don't have a newline in the input (so only a single line as input), the result will be a about 2% slower because of the extra tail.  What I mean is probably measured in hundreds of a second.

· In the cases where you have multi-lines (for example 1000 lines), you will see at least a 7000% performance gain.

So don't worry about the tail, if you don't know what your input is.

Aren’t One-liners even faster ?

Some people think that writing one-liners must definitely be faster than multi-line code.  Let’s compare the two

last_line = tail(mytext,1);

last_word_list = ntharg(last_line, "1-");

last_word = tail (last_word_list,1);

and

tail(ntharg(tail(mytext,1),"1-"),1);

If you run both pieces of code through the standalone psl compiler (psl –q) you will get :

last_line = tail (...)

last_word_list = ntharg (...)

last_word = tail (...)

and  

@temp_1 = tail (...)

@temp_2 = ntharg (...)

@temp_3 = tail (...)

At a first glance, you will say that this is the same, but if you look carefully there is more to say than that.

For the one-liner, the PATROL agent has assigned unnames local variables himself.  These are temporary variables that he needs in order to pass arguments.

These unnamed variables are @temp_1, @temp_2 and @temp_3.  In the first dump, you can see that these unnamed variables have been replaced with my “named” variables.

These are : last_line, last_word_list and last_word

So there is definitely no performance gain, but it’s even worse if you use one-liners.

If you would have liked to use the result of the ntharg() for something else (eg. To get the first word of the last line), you wouldn’t have been able to access that data, since all @temp variables are internal to the interpreter and cannot be used in PSL.

And … one-liners can definitely be harder to maintain.

Don’t trust everything you copy and paste

A typical PSL developer makes extensive use of the copy/paste buttons.  Sometimes that is a good thing, but very frequently the source you are copying from might not be written in the best way.

For example the PSL reference manual is supposed to be educational material and the majority of the examples in the PSL reference manual are good examples.  From time to time, you will run into an example that was written not suited for what you are trying to do.

You must understand that the examples are there to explain how the function works, not to give people the fastest code of doing things.  Readability is also very important in a manual that shows code samples.  You also have to keep the examples simple.

From what we learned so far is that optimization doesn’t necessarily means complex, but sometimes introduces some complexity in the thought process.

The most important thing is : you have to think for yourself as you write

Just to proof that some things can be done better, I looked in the PSL manual…

On PSL manual page 4-48 you will find an example that will display this :

Waveform for consine :

     -1 -0.5   0   0.5   1

     +----+----+----+----+

0.00           |         *  cosine = +1.000000

0.10           |        *   cosine = +0.995004

0.20           |        *   cosine = +0.980067

0.30           |        *   cosine = +0.955336

0.40           |        *   cosine = +0.921061

0.50           |       *    cosine = +0.877583

0.60           |       *    cosine = +0.825336

0.70           |      *     cosine = +0.764842

0.80           |     *      cosine = +0.696707

0.90           |     *      cosine = +0.621610

1.00           |    *       cosine = +0.540302

1.10           |   *        cosine = +0.453596

1.20           |  *         cosine = +0.362358

1.30           | *          cosine = +0.267499

1.40           |*           cosine = +0.169967

1.50           *            cosine = +0.070737

1.60          *|            cosine = -0.029200

1.70         * |            cosine = -0.128844

1.80        *  |            cosine = -0.227202

1.90       *   |            cosine = -0.323290

2.00      *    |            cosine = -0.416147

2.10     *     |            cosine = -0.504846

2.20     *     |            cosine = -0.588501

2.30    *      |            cosine = -0.666276

2.40   *       |            cosine = -0.737394

2.50  *        |            cosine = -0.801144

2.60  *        |            cosine = -0.856889

2.70 *         |            cosine = -0.904072

2.80 *         |            cosine = -0.942222

2.90 *         |            cosine = -0.970958

3.00 *         |            cosine = -0.989992

3.10 *         |            cosine = -0.999135

3.20 *         |            cosine = -0.998295

The developer who wrote this, obviously didn't think of all the functions you have at your disposal in PATROL, but probably wanted to make to code readable.

This is his listing :

function main() 

{


print("Cosine Waveform:\n\n");


print("    -1  -0.5   0   0.5   1\n");


print("     +----+----+----+----+\n");


i = 0.0; 


while (i <= 3.2) 


{



printf("%4.2f ",i);



cosine = cos(i);



plot = int(10 * cosine) + 10;



if (plot < 10) 



{




before = "";




j = 1; 




while (j <= plot) 




{





before = before . " ";





j++;




}




after = "";




j = plot; 




while (j < 9) 




{





after = after . " ";





j++;




}




print(before,"*",after,"|");




printf("            cosine = %+8.6f\n",cosine);



} 



elsif (plot == 10) 



{




print("          *           ");




printf(" cosine = %+8.6f\n",cosine);



} 



else



{




print("          |");




before = "";




j = 11; 




while (j < plot) 




{





before = before . " ";





j++;




}




print(before,"*");




after = "";




j = plot + 1; 




while (j <= 22) 




{





after = after . " ";





j++;




}




print(after);




printf("cosine = %+8.6f\n",cosine);



}



i = i + 0.1;


}

}

The logic of this program is to build each line character by character.

Personally, I don’t think the program is very readable.  If you understand what he does, it is OK, but if you have to change something, you must be careful where you change what ...

Also there is a lot of looping and branching.

Thinking it all over, I came up with the following listing :

# Print the header

print("Waveform for consine :\n\n".


"     -1 -0.5   0   0.5   1\n".


"     +----+----+----+----+\n");

# We know each line will at least print this character string

# plus an asterix somewhere

whole_line="          |          ";

# Loop over the degrees

for (i=0;i<=3.2;i+=0.1) 

{

  # Find the cosine value

  val=cos(i);

  # See where on the line of 20 long the asterix should be put

  plot=int(10*val)+10;

  # Break the line in two and later rebuild the line 

  # using a printf

  part_one=substr(whole_line,1,plot);

  part_two=substr(whole_line,plot+2,20);   # 20 is maxlength of line

  printf("%4.2f %s*%s cosine = %+8.6f\n",i,part_one,part_two,val);

}

The logic of this program is to take a line and put the "*" where it belongs...

We just break the line apart where needed.

This increases the speed incredibly.... (x2) and it (to my understanding) more readable code.

Another approach might have been that you took the first bit of code, and rewrote it so it looked more "compact"  (we already know this isn’t necessarily good).

Let's take a look at what we can do when we look at the first example,

but rewrote it a bit :

for (i=print("Cosine Waveform:\n\n    -1  -0.5   0   0.5   1\n     +----+----+----+----+\n");i<=3.2;i+=0.1) { printf("%4.2f %s cosine = %+8.6f\n",i,(plot = int(10 * (cosine = cos(i)))) < 0  ? sprintf(sprintf("%%%ic%%%ic|           ",11+plot,-plot),"*","") : ((plot == 0)  ? "          *           " : sprintf(sprintf("          |%%%ic%%%ic",plot, 11-plot),"*"," ")),cosine);}

This example IS a huge one-liner in a loop and it does the same.

It uses advanced functions like nested sprintf's  and nested ternary operators, it also uses assignments in statements and every other thing you wouldn't do as a beginning programmer.

As a result, the code we produced is unreadable, unmaintainable and even not better/faster than the optimization we did.

Creating instances (the create() statement)

Mostly the first KM’s you write contain badly written discovery scripts that are not very readable and lead to poor performance.  Usually writing good discovery code is something you learn over time (by trial and error).

Classic Create Loop

Somehow, discovery and pre-discovery has always been something that seems difficult to write.  The reason for this is the direct result of the traditional way of development: first make sure it works, then make sure it works well.

This means that you first start working on the create statements and later (when the product is tested), people find out that it is necessary to add destroy() statements to the code as well.

In many discovery scripts the create cycle is so big and sometimes so complex, it is not easy to add destruction code, the result is code that is bigger and more complex.

There is a general flow that can be recognized in these KM’s

First, we need a way of determining the instances we want to create for the application.  For this example, we will say that we will store this in the list : wanted_instances

In a lot of KM’s you will see code that works like this

wanted_instances = <command that returns list of instances>

foreach instance (wanted_instances)

{

if (!exists(instance))


{



create(instance,instance,OK);


}

}

This code is OK, and works fine, but as a result you are checking for every instance in the list if the instance object needs to be created.  This means, performance of the code depends on the number of instances.  Or at least we can say, the more instances you have, the more CPU this will consume.

Furthermore every time discovery runs we will run over the same code again only to find out all instances have already been created.

The Classic Destruction Loop

If you add destruction logic to this application class, you cause a bigger impact on performance.

current_instances=get(“instances”);

foreach instance (current_instances)

{


found=0;


foreach good_instance (wanted_instances)


{



if (instance == good_instance)



{




found=1;



}


}


if (found == 0)


{



# This means an instance is in the current instance



# list but not found in the wanted instances



destroy(instance);


}

}

In this code sample we have created a loop within a loop for each discovery cycle. 

Let’s take the case that our application discovery is quite static.  This means that all instances have already been created and none will be destroyed.

Looking at the code, that means that current_instances will be equal to wanted_instances, and that “found” will always be 1.

If you take a close look at the code this means that we will have to do a lot of looping, just to find out nothing has to be done.

This creation “logic” therefore results in a lot of needless executions.  The “if-condition” is executed once for each item in the wanted_instances list.  The execution of the wanted_instances iteration is repeated for each item in the current_instances list (current_instances*(current_instances+1)).  In a KM with 31 instances, this would mean almost 1000 if-statement executions for every 45 second discovery cycle.

Sure we should be able to do better than this.

Optimizing The Discovery Process

We’ve seen that nesting foreach loops can result in large numbers of needless executions.  To optimize our code we eliminate the foreach loops and replace them with the very powerful difference() function.

The difference() function returns a PSL list with all the elements that are unique between two lists.  Rather than iterating through the full list everytime we execute a discovery script, we use the difference() function to process only those items that are new or not contained in both lists.

Lets look at this in code :

wanted_instances=<…command that will return list of instances…>;

current_instances=get(“instances”);

to_create=difference(wanted_instances,current_instances);

to_destroy=difference(current_instances,wanted_instances);

foreach instance (to_create)

{


create(instance,instance,OK);

}

foreach instance (to_destroy)

{


destroy(instance);

}

This code listing performs the same list comparisons in the previous listing but much faster.  The foreach loop is only called when there is actually a difference between the wanted_instances and current_instances lists signifying that an instance object needs to be created.  The code eliminates needless iterations and is much easier to read.

For each additional instance, there will be some extra CPU necessary for the difference call, but the impact is minimal as the difference function() is a very fast instruction.

You will only use more CPU if there is more work to be done (more instances created or destroyed).

Adding Return Code Checking

If you code properly, you should check for return codes of the different PSL functions.  One important code to check is the return code from create.  Don’t expect every create statement to succeed.

In some of KM’s you see code like this:

wanted_instances=<…command that will return list of instances…>;

current_instances=get(“instances”);

to_create=difference(wanted_instances,current_instances);

foreach instance (to_create)

{

   create(instance,instance,OK);

   data=<…command that returns inst specific data…>;

   set(instance.”/PATH”,data);

}

This code looks fine, but if a PATROL Administrator wants to use the built-in filter feature of PATROL to filter out instances using the

/AgentSetup/<APPL>.filterlist 

configuration variable, the create statement In the second foreach statement will fail and generate a runtime error.

The problem is that the programmer will not be aware that the create statement failed and will consider the object for the application instance created. Many downstream coding errors will result.

Therefore it is better to check if create() succeeded

Listing 1.4

wanted_instances=<…command that will return list of instances…>;

current_instances=get(“instances”);

to_create=difference(wanted_instances,current_instances);

foreach instance (to_create)

{

  if (create(instance,instance,OK))

  {

    data=<…command that returns inst specific data…>;

    set(instance.”/PATH”,data);

  }

}

Sometimes discovery acts as a collector as well, in that case it might be a good idea to split the collection logic from the create logic

Listing 1.5

wanted_instances=<…command that will return list of instances…>;

current_instances=get(“instances”);

to_create=difference(wanted_instances,current_instances);

foreach instance (to_create)

{

   if (create(instance,instance,OK))

   {

      data=<…command that returns inst specific data…>;

      set(instance.”/PATH”,data);

   }

}

# Requery current_instances for collection

# Create might have failed

current_instances=get(“instances”);

foreach instance (current_instances)

{

   set(instance.”/Status/value”,…);

}

Add instance filtering using filerlist

The PATROL Agent lets you define what objects are displayed by setting a configuration variable called:

/AgentSetup/<application_name>.filterlist 

This variable contains a comma-separated list of the application instances that you want to filter. (See the PATROL Agent Reference Manual for more information).  This is probably the cheapest way to do filtering when you develop your KM.

To change the value of this configuration variable you can use any of these pconfig() PSL calls.

pconfig("APPEND","variable","value");


adds “value” to “variable” 

(even if “value” already exists in “variable”)

pconfig("DELETE","variable","value");


if (“value” exists in “variable”)

deletes “value” from “variable”

pconfig("MERGE","variable","value");


if (“value” doesn’t exist in “variable”)

adds “value” to “variable”

pconfig("REPLACE","variable","value");


sets “variable” to “value”


pconfig("DELETE","variable");


deletes “variable” from the configuration database

pconfig("GET","variable");


returns a list of



default value\n



current value

Note : Using get(“variable”) on a config variable will also work, this will only return the current value.  The cost of get() or pconfig(“GET”,…) on a configuration variable will be the same.

An example to filter out an instance using a menu command:

appl_name=get(“../name”);

inst_sid=get(“sid”);

pconfig(“MERGE”,”/AgentSetup/”.appl_name.”.filterlist”,

inst_sid);

destroy(“.”);}

If this code is pasted in a menu command "Filter Instance", then when it is selected, the current instance will be filtered out, using the filterList and the instance will immediately be destroyed.

You will have to make sure that there is a way to undo this filtering, if you filter out all instances, you will have an empty application list (A useless KM).   Therefore some KM’s check if the instance is the last instance they will remove.  If that is the case, they first create an OFFLINE setup icon.

Introducing Create-Condition Check

In some cases, discovery is completely dependent on the content of a file.

Normally this file needs to be parsed to create the “wanted_instances” list.

A good way to program this is first to check if the file has changed.  If the file has changed, you can parse it and see if you need to add/destroy instances

file_to_check=<…set this to the filename that should be checked…>

timestamp=file(file_to_check);

old_timestamp=get(“old_timestamp”);

if (timestamp != old_timestamp)

{

   if (timestamp)

   {


      #now process the file and do your actual discovery

   }

   else

   {


#process “file doesn’t exist” (anymore)

   }


   # Make sure to save the new timestamp

   set(“old_timestamp”,timestamp);

}

The first time discovery runs, get(“old_timestamp”); will return NULL (and an “invisible” runtime error).

That means that timestamp and old_timestamp will always be different.

(in fact that is not really true : if the file doesn’t exist, file() will return NULL as well, and in that case it will be the same as old_timestamp. this is not a problem, because we won’t process the file if it doesn’t exist)

If the file has been processed, the timestamp will be saved.  On the next discovery cycle we will look for the old timestamp again (the one we saved earlier) and use that timestamp to check if the file changed.

Limiting Number of Created Icons

When building a KM, it is always important to think about the consequences that PATROL will have on your host OS.

It is not always easy to find relations between PSL code and CPU of the agent, but it is pretty easy to see that the agent’s footprint (memory, CPU) will be directly related to the number of APPLICATIONS, INSTANCES and PARAMETERS.

Limiting the number of application classes will have a limited impact on the Agent’s CPU. The Application class itself uses some memory and some CPU (Prediscovery/Discovery).

Where it really starts getting important when you start instantiating instances of the application.  This can be easily demonstrated by the following example.

Let’s say we have 4 KM’s that are nested.   

KM1 has 15 instances

When you click on an instance of KM1, you see 15 instances of KM2

When you click on an instance of KM2, you see 15 instances of KM3

When you click on an instance of KM3, you see 15 instances of KM4

Let’s say that KM4 has 2 parameters, each with a schedule of 1 minute.  All the other KMs don’t have any parameters.

What is the total number of parameters we will execute every SECOND?

2
parameters/KM * 15 instances of KM4
= 30 parameters

30
parameters/KM4 * 15 instances of KM3
= 450 parameters

450
parameters/KM3 * 15 instances of KM2
= 6750 parameters

6750
parameters/KM2 * 15 instances of KM1
= 101250 parameters

Dividing the total parameters by 60 (seconds) yields 1687 parameters/second.

or 60,750,000 parameters every HOUR (101250 * 60)

You see it is important to limit the number of instances (and certainly when you write nested instances). 

So when creating a lot of instances, think !!  You might consider filtering of some sort so the end user will not be overwhelmed by the amount of instances.

Here are some basic filter methods you could consider

· Top X
Only monitor the top X processes. 

· Only show NOT OK instances
This is really useful for user monitoring.  The operator might not care about certain users unless they are in trouble. This type of monitoring is also called “Conditional Monitoring”.  The operator should be able to set the conditions.

“Monitor Specific Instance” can extend this so that an operator can add specific instances unconditionally.

· Limit number of instances.
In certain cases it would be a good idea to add a limit to the number of instances.  Just in case there is a problem affecting all users.

· Use Exclusion List.

· Filter By Simply Not Showing
Sometimes instances are created, and never removed. Or sometimes instances are merely used as a type of “online report”.  Consider removing the instances and create some real reports as menu commands.

· Create summary (less detailed) view
Using the example of the nested KMs, move the information of KM4 to a menu command under KM3.  On KM3, create summary parameters of the parameters previously shown under KM4…

Provided you would still meet the functional requirements for the KM, changes of this type will reduce the footprint of the agent (in the case of just removing KM4 in the example, the improvement would be 95%).

PATROL Agent

Should I log messages in the “Agent Error Log”?

The agent error log is indeed "The agent error log" and not "km error log".

Some developers write severe errors to the system output window, but even in that case you should provide an option to turn this off, because even if the errors are severe... some people will say it's a kind of debug info and one should be able to disable debug.

BMC don't offer access functions to the agent error log just for the reason that it is not a KM error log... However, BMC also doesn’t offer a KM error log.

What you can do for example is to trigger an event (that is if you don't want to worry about file cleanup, maintenance, etc...) otherwise, just maintain your own logfile.

How can I open a visible window on a PATROL agent?

This question only applies to agents running on Windows NT/2000

Sometimes people want to open a visible window (GUI process) on a PATROL agent (for some bizarre reason).  Since the agent is running as a service, with no console context, it is sometimes considered impossible.  Actually when you do a system() function on the agent, you will see the process in the taskmanager, but the GUI won’t appear.

A simple solution to that problem is to call the GUI process via another executable that will have a detached window context.

Source code for such an executable can be found below.  A compiled version can be found on http://devcon.bmc.com
#include <windows.h>

STARTUPINFO si; 

PROCESS_INFORMATION ProcessInformation; 

void

main(int argc, char **argv)

{

  if (argc < 2)

  {

    exit(1);

  }

  si.cb = sizeof(STARTUPINFO); 

  si.lpReserved = NULL; 

  si.lpTitle = NULL; 

  si.lpDesktop = "WinSta0\\Default"; 

  si.dwX = si.dwY = si.dwXSize = si.dwYSize = 0L; 

  si.dwFlags = 0;; 

  si.wShowWindow = SW_SHOW; 

  si.lpReserved2 = NULL; 

  si.cbReserved2 = 0; 

  if (CreateProcess(NULL, argv[1], NULL, NULL, FALSE, 0, NULL, NULL, &si, &ProcessInformation)) 

  { 

    CloseHandle(ProcessInformation.hProcess); 

    CloseHandle(ProcessInformation.hThread); 

  } 

}

Can I make my KM preloaded from PSL ?

If you execute the following PSL function:

pconfig("MERGE","/AgentSetup/preloadedKMs","<KM TO PRELOAD>");

The <KM TO PRELOAD> will be preloaded on the agent.

If this fails, then you probably have a permission problem (take a look at the /AgentSetup/accessControlList). You will have to investigate what is in there, but you might need configuration privileges for the patrol account on the local box.

Adding an entry like "patrol/localhost/C" will do the trick!

(Agents later than 3.3 don’t need this ACL setting anymore).

How can I check for a remote agent status?

Let’s presume you can't or don’t want to use remote_open, because you feel there must be better ways to check remote agent availability (in which case you are right).

You could use "PatrolCli", an executable that is shipped with every PATROL installation.

PatrolCli does a very lightweight session layer level ping, and it has some nice features like scanning a whole subnet for agent availability at once.

The help for PatrolCli ping is:

ping [-subnet] [-n] HOST [PORT] [SECS] - check if Patrol Agents responds

HOST is a hostname or IP address

PORT is a number or range N1-N2

SECS is an the timeout value (default is 5 sec)

-subnet means check all hosts on the subnet on which the IP address represented by HOSTS resides

-n means no address to hostname resolution 

Examples:

ping myhost 2200-2220 2 : check if there is PatrolAgents on "myhost" on ports 2200...2220, 2 sec. timeout

ping -subnet 284.12.55.0 : check for PatrolAgents on default port (3181) on IP addresses 284.12.55.*

ping -s -n hudson "" 1 : scan the C-class subnet that includes "hudson" output IP addresses instead of host names, use the standard port and timeout 1 second

For further info on PatrolCli, you should read the manual "PATROL Command Line Interfaces Manual", that you will find on your PATROL documentation CD.

If you want you can call PatrolCli from PSL like this

system("PatrolCli \"ping myhost\"");

Or better, you could create a command type PATROL_CLI that calls PatrolCli and call it like this:

execute("PATROL_CLI","ping myhost");

How do I inform another agent a certain event has happened?

An example would be:”If a process being monitored on a remote machine goes down what is the best way to inform another agent on another machine that the process went down?”

You can let the agent that finds out the process has died push the notification to the other host, or you can write a process that will query the other host if a process died. 

Pushing the information

* SNMP trap : You can send an SNMP trap to the other system using snmp_trap_send(). On the other side you will need a PSL process to listen from incoming traps using snmp_trap_listen().

For this method you would need to know the destination host and community strings.

* SNMP agentExecuteCommand: You can use SNMP as well to execute a remote PSL command. You will have to do and SNMP set on the MIB variable agentExecuteCommand.

For this method you would need to know the destination host and community strings.

* Remote Agent Communication: you can do a remote_open() and trigger and event on the remote system. This event can execute a function on the remote agent.

For this method you would need to know the destination host and a valid account that is allowed to connect to the agent.

* PatrolCli: You can use PatrolCli to connect to the agent and trigger and event or execute PSL on that remote system

For this method you would need to know the destination host and a valid account that is allowed to connect to the agent.

* (D)COM: If the agents are running on NT, you can write a small application that will use the PATROL COM interface to execute or set something on the remote system.

For this method you would need access permissions to the PATROL agent com server which means: COM must be configured properly

* PEMAPI: You can write your own PEMAPI program that will connect to the remote agent and sends and event or executes a PSL script on the agent

For this method you would need to know the destination host and a valid account that is allowed to connect to the agent.

* Various other methods:

There are lots of other methods available outside of PATROL. Some of them are

- Sending an email and have an email processor on the other end

You don't necessarily need to know anything but the email address here

- Just putting a file on a shared file system and the other system will pick that up

You will need to know the name of the shared file system

- Have a small client server program that allows you to pass information to the other side

Pulling the information

In this case, the agent that finds out that the process is missing will have to set a parameter in alarm.

* SNMP: at regular intervals you can do an SNMP get operation to see if the parameter is still OK.

* Remote Agent: Now the central side will contact the hosts and ask if there is something wrong (This method is probably not very interesting).

* (D)COM: Same thing as the remote agent communication but over DCOM (NT Only)

* PEMAPI: Same procedure (both UNIX and NT)

Receiving through the Event Mechanism
The host that finds out something goes wrong just triggers an event in his private event catalog with all the explanation that the receiver would need.

The receiver side has a process running that will connect to the agents using PEMAPI and will look for events from that specific event catalog.

Whenever a process dies, the triggered event will be received by that program and it can then tell the agent who needs to know.

This requires some programming and you will have to know usernames and passwords as well.

This list is probably far from complete, and there may even be better alternatives, I just wanted to give a couple of ideas...

When/How does history expire?

When history expires, what happens to the data in the param.hist file? Let's say you have a filesystem instance that exists for only a few moments and is then unmounted or excluded. Would the history for the parameters of that instance persist in the history file forever? Does data ever get "erased"? Can the history file go down in size?

You could think of the PATROL History database as a "Circular" file, where the latest data is written, and the expired data is written over. The history database could increase in size if you added other parameters/application classes or increased the retention period of any of your parameters.

But then what about reducing the file size?  The problem is that one would like to achieve certain effects with custom KMs without blowing up the history file. This is proving rather difficult. For example, occasionally bringing in unique and temporary filesystems would slowly bring the size of the history file up. There would be no certain limit.  Is there a way to store history on continually changing instances without having the history file run away?

This description of the problem is commonly known as "history pollution".

There is a method you can use to work around this, but it might not be acceptable for all cases.

Let's say you would like to monitor processes and display the combination of process name + PID and maintain some history about them.

This will be a very good example to get maximum history pollution, since the process name and PID is an extremely unique thing, and it's unlikely that this combination will ever reoccur once the process has died.

It's important to know that the history uses the combination of /APPL/SID/PARAM as an index to store the historical data.

APPL = application class name

SID = instance id (or sid in the namespace)

PARAM = parameter name

What the end user will see is on the console is not the instance ID, but the instance label (which is the "name" in the namespace), and that is something we can use.

Just think of an instance ID as a “slotnumber” and the instance label as the thing we want the end user to see.

Applied to our processes example, you could come up with something like this:

/APPL/SID=

- /PROCESS/SLOT1

- /PROCESS/SLOT2

- /PROCESS/SLOT3

but the visible instance name would be

/APPL/INST=

- /PROCESS/inetd-445

- /PROCESS/ksh-336

- /PROCESS/xterm-776

As long as the number of processes will be reasonable, we will not pollute the history more than necessary. (We will only create to number of slots we need, and reclaim them whenever one becomes available)

Of course this feature might sound very good, but there is a problem with this approach as well.

If process A was using slot 1 and it dies, it becomes available for a new discovered process, let's say process B.

If we now ask for history of a parameter belonging to process B, we will not only get the history from the B process, but also the old history of the A process (or even every old process that occupied the slot in the past).

That means we have to provide addition info on the graph, so we won't confuse our end user what information he's actually looking at. One way to do that would be to annotate every time a process occupies slot and when it releases the slot.

Another way would be to define a certain "impossible" value for each of your parameters and set the parameter to this impossible value whenever a switch occurs, or whenever no one is occupying the slot.

Of course it would also be possible to use a combination of both methods explained above.

Maybe it would be better to go into the details of how the agent stores history.

The history file uses blocks to store data for parameter/instance values. Over time, blocks are freed (because of history retention) for use with new data, but the new space allocated will only be for values of the same parameter/instance. In other words, if you have an instance /FILESYSTEM/tempfs01 and store history on it, when the history expires, there will be room allocated in param.hist for /FILESYSTEM/tempfs01 (only) even if that instance does not exist anymore. There is currently no direct method in PSL to handle this problem.

This is certainly different than how most people think it works.  Usually, one assumes that once the history for /CLASS1/INST1/PAR1 was "out of date" then the space it took was available for any other /CLASSx/INSTx/PARx to fill.

Actually it is quite logical that this is not the case.  The history file works just like a database. It is indeed the case that whenever history is outdated, the occupied space will become available again for other /CLASSx/INSTx/PARx values.

The problem lies somewhere else …  when you set history retention period to 7 days, that means the oldest datapoint for a certain /CLASSx/INSTx/PARx will not be removed from the history database if the timedifference between the oldest datapoint and the newest datapoints is less than 7 days.

In case INSTx is reasonably unique (for example, "PID-PROCESSNAME" combination), and the history retention period is 7 days... you will eventually end up with a huge history file, because after the process goes away, the history will remain.

Maybe another example is better: Let's say we have a process that restarts every 2 days. That means every two days it will be assigned another PID. The data in the history database will not be cleaned up until there are 7 days worth of history. That means the index and data in the history database will never be removed.

Best case scenario, the process continues to run and will "only" occupy 7 days worth of data, since the history works like a round robin database data which is older than now-7days will be overwritten with new data.

Whenever the process dies (and the instance is removed), your history data will not be cleaned up, because the rule is to keep 7 days worth of data … so it's sitting there and you can't access it.

For most KM's this is OK, for example a filesystem is unmounted, instance is removed, filesystem is remounted, instance is recreated, and the "old" data is still available.

However, with our example, it is very unlikely that the combination PID-PROCESSNAME will ever reoccur... and your data will just be sitting there.

Since no extra data is added, there is no reason for the agent to clean it up (as mentioned before, history will be cleaned when is the delta between timevalues of old and new datapoints are greater then the history retention period).

Of course it doesn't matter how many times/year the process restarts, it is important to know that the agent will store (and not necessarily free) data for each /CLASSx/INSTx/PARx combination. The only freeing process that happens is either console triggered (developer console can call "Clear History"), or automatically when the timerange of a parameter exceeds the history retention period.

PATROL Console

Can I use PSL on the console?

Only the PATROL Agent executes PSL.

That means if you install an agent on the same machine a console is running, you can let that agent do some PSL for you.

Also the standalone PSL interpreter can be used on the console, although not all PSL functions will be available to you - like getting data from the namespace for example or calling system() or execute().

How do I launch a program from the console?

There is one way to execute menu commands locally on the console.

Create a menu command of command type OS

As the first line in your menu command script type:

#%MODES% local

If the console sees this line in the menu command code, it will execute the code on the console.

On NT you can do the following

#%MODES% local

c:\\mydir\myfile.htm

Depending on the default settings of NT, this will launch either Netscape, Internet Explorer or any other default browser. (Of course if c:\mydir\myfile.htm exists)

You can specify some macros.  Here is a list of supported macros:

	APP_TYPE
	Application name (e.g., ORACLE)

	COMP_STATUS
	Computer’s state (e.g., VOID, OK, WARNING)

	COMP_TYPE
	Computer type (e.g., SOLARIS)

	CON_INFO
	Network connection information (e.g., host name or net address)

	CON_STATUS
	Network connection state (e.g., OK, Soft Problem)

	CON_TYPE
	Network connection type (UDP or TCP)

	DISPLAY_NAME
	Host name where the console is displayed

	HOME
	Home of this parameter, task, or command

	HOSTNAME
	Host name

	INSTANCE_ID
	Instance id (e.g., ORACLE SID)

	LANGUAGE
	The language to use in sound

	PASSWORD
	User’s password

	SID
	Instance id (e.g., ORACLE SID)

	STATUS
	Status of computer or task

	TIME
	Current time of day

	USERNAME
	User’s login name

	VOLUME
	Sound volume

	WORST_CHILD
	Worst child (parameter or instance) running on this object


Use these macro commands like you would use any macro command in PATROL... for example %{SID}

How do I get a list of all hosts defined in the PATROL console?

Although very unsupported, this code does the trick

# Change this for your situation, we are looking for the file : 

# computer-[session#].idx

consolesession=1;

localdir="c:\\patrol\\";

ok_lines=cat(localdir."computer-".consolesession.".idx");

ok_lines=trim(ok_lines,"\r"); # Good practice for portability

ok_lines=ntharg(ok_lines,1,"|");

ok_lines=ntharg(ok_lines,2,"/");

ok_lines=grep(".*:$",ok_lines);

hosts=ntharg(ok_lines,1,":");

print("The defined hosts in the console definition file are :\n\n");

print(hosts);

How do I execute a multi-line PSL script from PatrolCli?

Sometimes people want to execute multi-line PSL scripts from PatrolCli and shortly they find out that only the first line is executed (the rest seems to be ignored).

Let’s say you want to execute a PatrolCli command that will return all loaded applications.  Code to do that is:

applications=get_vars("/","subnodes");

foreach appl (applications)

{

  applpath="/".appl;

  if (get(applpath."/__type__") == "APPLICATION")

  {

    ret_data=[ret_data,applpath];

  }

}

# Now we want to return ret_data …

When you would trace the remote PSL process that is launched, you would see that actually all lines will execute, but you will only get the result of the first line. Which is:

applications=get_vars("/","subnodes");

The reason for this is that your execpsl is actually being executed in the RemPsl standard event's notification action.

The code in the RemPsl standard event notification action is basically:

# This command assumes that event arguments are separated by tab

xxx_pem=%{EV_ARG1}

event_trigger("0","Result","INFORMATION",

              "6",xxx_pem,"%{EV_ARG2}");

The first line in the RemPsl standard event notification action will cause the execution of your script since a replacement of %{EV_ARG1} will occur with your PSL script before compilation and execution.

This will thus result in

xxx_pem=applications=get_vars("/","subnodes");

foreach appl (applications)

{

  applpath="/".appl;

…etc.

that means that only the result of first line will actually be passed back through the event_trigger().

A way to execute multiple lines and get the result of what you want it to be is by assigning the result to xxx_pem yourself in your script.  That means the last line should be :

# Now we want to return ret_data …

xxx_pem=ret_data;

Another (simple) example to this would be

a=”hello”;

xxx_pem=”a.world!”;

Could be executed like this :

execpsl "0;\na=\"hello\";\nxxx_pem=a.\" world!\";\n"

And will result in the following script to be executed in the notification action:

xxx_pem=0;

a="hello";

xxx_pem=a." world!";

event_trigger("0","Result","INFORMATION",

              "6",xxx_pem,"%{EV_ARG2}");

PSL Language

How can I trace a PSL process?

How can I detect PSL runtime errors?

The easiest way to detect runtime errors is by setting the PslDebug variable.

This is a magical variable inside the PSL interpreter that will determine the verbosity of the execution.

If you set PslDebug to 0, almost all runtime error messages will be suppressed and no execution trace will be shown.  If you set it to –1 the interpreter will be very verbose.

Here is a list of possible values you can assign to PslDebug (the default value of PslDebug is 32).

For example:

PslDebug=-1;

print(“this call is being traced”);

PslDebug=0;

print(“all runtime checking is turned off”);

	Value
	Definition

	-1
	Enable all PSL run-time error checking and tracing functions.

	0
	Disable all run-time error checking and tracing.

	1
	Enable all PSL run-time error checking.

	2
	Enable warnings when arithmetic operations involve NULL string operands.

This check is a stylistic one used to verify that numeric variables are explicitly initialized to zero rather than defaulting to NULL strings, which are treated as zero in arithmetic operations.

	4
	Enable warnings for arithmetic operations involving nonnumeric operands.

	8
	Enable warnings for illegal or undefined arithmetic operations.

	16
	Enable warnings for variables that were not explicitly initialized. 

This checks for variables that are used before being explicitly equated to a value.

	32
	Enable warnings for PSL Version 2.0 built-in functions.

	64
	Enable warnings for PSL Version 3.0 built-in functions.

	256
	Enable all PSL run-time tracing.

	512
	Enable function call tracing.

Function call tracing reports which functions are called but does not return

information about the arguments. Function call tracing traces both user-defined and built-in functions.

	1024
	Enable function argument tracing.

Function argument tracing reports the arguments passed to all user-defined or built-in functions. Function argument tracing requires that function call tracing (PslDebug = 512) also be enabled; that is, PslDebug = 1536 for function argument tracing.

	2048
	Enable function return value tracing.

Function return value tracing reports the value returned by calls to all user-defined or built-in functions.

	4096
	Enable variable assignment tracing.

Assignment tracing reports the variable name (if available) and the value

Assigned to it.

	8192
	Enable errno tracing.

errno tracing reports any nonzero values stored in the PSL errno variable.

	16384
	Enable PSL lock tracing

Lock tracing reports the interprocess actions that occur during lock() and

unlock() function processing, including the granting, denying, and releasing of locks.


How do I work with PSL binaries?

Any PSL file that will be run by the agent will first be compiled into bytecode (quadcode).  It is possible to run this precompilation step so the agent won’t have to do this (can be useful for source code hiding for example).

If you have a syntactically correct PSL script, run it through the standalone compiler/interpreter and create a .bin file.

For example:

psl –b mycode.psl

Will create a file “mycode.bin”.

If you would ship this file to the agent instead of the PSL file, the agent will load it directly without compiling the code first.  (in case a psl file is present, the agent will load the psl file instead and compile the PSL file, instead of relying on a possibly outdate bin file).

How do I embed other languages in PSL?

A unique feature of PATROL is that you can incorporate scripts you may have already created simply by calling them, either directly as system scripts or via their interpreter command.

Is accessing the config database an expensive operation? 


You have to understand that every pconfig command will physically access the configuration database! This is not really a very cheap operation (certainly if you will be doing it a lot).

Can I set a parameter value retroactively?

There is a way to get this done, however unsupported, undocumented and it should be used with extreme caution...

There is a 3rd optional argument to the set() function. This argument will accept the timestamp (in PATROL time-format = seconds since 00:00:00 GMT, Jan 1, 1970).
That means, you can do the following:

set("/APPL/INST/PARAM/value",<value>,<timestamp>);


Use with caution: 
New set() function calls should always have a newer timestamp (higher value) than any previous set() call !
That means ... never set() something back in time if you already have already done set() for more recent datapoints.

NOTE : This is unsupported by BMC and the functionality might go away in future versions, however if you need the functionality now... it might be a useful "feature".

How does the escape character ‘\’ work?

The single '\' character is an escape character that will be interpreted at compile time of the PSL script. That means, even before the string is stored internally.

There are only a few string literals supported:

\t tab

\n new-line

\r return

\b backspace

\A..\Z Ctrl-A ... Ctrl-Z

\\ escaped backslash

\" escaped quote

Sometimes people wonder if this can be “undone” to show the real entered string in a text window for example.

In some cases it is impossible to “undo” the operation … the easiest way to show that is the following :

%PSL print(length("AAA"));

returns 3

%PSL print(length("\t\t\t"));

returns 3

BUT

%PSL print(length("\p\p\p"));

returns 0

That means, PSL doesn't even know those characters were there, because they are unknown string literals!

Of course, the question is now: how do I prevent that ?

You say you want to recover when a user enters a \ instead of a \\.

There are (to my knowledge) only two ways for a user to enter information and I'll explain the consequences of each of the possibilities

Using the %%{} macro 

This is a "console side" feature. That means every %%{} macro will be expanded before the command is send to the agent. Actually the agent doesn't know (or need to know) about this macro at all. It will just receive the macro expansion instead of the macro.

If you use this method for getting user data, you have to make sure the user always specifies \\ instead of \.

Mistakes here can result in untrappable errors!

Therefore this mechanism is not really adviced to get user input.

In the following example (put this in a menu command)

user_input="%%{input :}";

If the user enters c:\patrol

The agent will receive the command as

user_input="c:\patrol";

And therefore you just introduced and untrappable mistake, because \p is an unknown string literal!

Using the response() function 

This is an agent side function. Anything entered in the response box will be returned literally. That means if a user enters "c:\test", it will be stored as "c:\test”!

This means it's automatically escaped (actually it's not, it's just literal...the escaping functionality is just disabled, and therefore it won't be interpreted as an escaped character)

Even if a user enters "one \n two" it will be returned literally. If you want to interpret the backslashes yourself, go ahead.

Just type:

%PSL print(response("Backslash tester","","","11\n"));

and enter something...you will see that exactly the same string is returned.

Note 

It might also be important to understand that this backslash interpretation is only done for literal strings.

That means

x="one \\n two";

print("X=".x."\n");

# this prints one \n two

y=x;

print("Y=".y."\n");

# this prints the same

# and not :

# one

# two

Will work as expected, you don't have the quote the backslashes anymore.

This backslash interpretation is actually done when the string is stored in the data region of your compiled PSL (at compile time), not at runtime... but that would lead us to far from the subject.

How do I enforce only a single instance of a response box?

Let’s say you only want to allow one operator at a time to open a specific response box.  That means you will have to use some sort of locking to make sure that the second response function cannot be opened.

If you use locking (the real lock() function), you will see that the agent will automatically release all locks held by the PSL process the moment it dies, so it seems that the PSL lock function will be able to help us out on this.

A piece of PSL code to illustrate that:

if (lock("MYLOCK","x",0))

{

print("I've got the lock\n");

}

else

{

print("Lock already held by someone else\n");

exit;

}

response("LOCK TEST","","","1\nCLICK TO RELEASE THE LOCK");

You will see this will work and will allow you to have only one response function of a specific type active.

But by doing this you have created another problem...

Let's say that your customer is using PATROL for 24h support and they have two operator sites (one in Belgium and one in Houston), so they can do "follow the sun" support.

A problem happened in Houston during daytime and someone opened the response function....  Now what will happen if the guy who opened the response function forgot to close it?

12 Hours later the same problem occurs. The guy in Belgium tries to open the response function and gets a message that someone else already opened it! 

In order to solve the problem (and get access to the response function), the guy in Belgium has to call the Houston office (probably no answer) or restart the agent (drastic intervention, just to release the lock)

So what do you probably also need here? An emergency unlock feature!!

I mentioned that a lock would automatically be released when a PSL process is killed, so we change the PSL script to the following:

if (lock("MYLOCK","x",0))

{

print("I've got the lock\n");

set("lockpid",getpid());

set("locktime",time());

}

else

{

lockpid=get("lockpid");

locktime=get("locktime");

print("Lock already held by pid ".lockpid." timestamp=(".locktime.")\n");

# Probably you would like to ask the user here if he 

# want to use the emergency

# unlock... we will presume he wants to

kill(lockpid);

print("We killed the process holding the lock, ".

      "try again\n");

exit;

}

response("LOCK TEST","","","1\nCLICK TO RELEASE THE LOCK");

Indeed, it seems the process can't be killed.... Well, this is something special; the process is in an IOWAIT state (see %PSLPS) and is therefore not considered to be running.

So how can we fix that problem?

Well, it is obvious that we won't be able to kill the response function (because of IOWAIT state) if we write it like that. Maybe we can rewrite the response function so we can kill the PSL process?

Take a look at this code:

if (lock("MYLOCK","x",0))

{

print("I've got the lock\n");

set("lockpid",getpid());

set("locktime",time());

}

else

{

lockpid=get("lockpid");

locktime=get("locktime");

print("Lock already held by pid ".lockpid." timestamp=(".locktime.")\n");

# Probably you would like to ask the user here if he 

# wants to use the emergency

# unlock... we will presume he wants to

kill(lockpid);

print("We killed the process holding the lock, ".

      "try again\n");

exit;

}

#Open a dynamic response function

rid=response("LOCK TEST",

             "","D=1","1\nCLICK TO RELEASE THE LOCK");

output="";

while (!output)

{

output=response_get_value(rid,1);

sleep(1);

}

# Output received, kill it

response(rid,"","K=1");

This will work, because we have created a while loop that will allow the PSL process to come out of the IOWAIT state once every second... this is enough to have the process killed (and release the lock in case of emergency)!

There are other workarounds possible to this problem.

What you can also do is maintain a list (PIDLIST) with PID's that started this menu command (and will potentially modify something in the response function) ... If someone saves the info, you check if his PID is (still) in the PIDLIST.

If this is the case, you save the info and clear out the PIDLIST...

This will prevent anyone else from saving the info (because the PIDLIST is cleared)

If someone opens the response box after the info was save, his PID will be added to the list again and he will be able to save his info.

I don't know if this procedure is clear, but it is NOT based on locking AND it is secure, because the list will only be cleared if someone saves the info (meaning that we don't care about processes that die/disconnect)

How do I save and use application passwords in PATROL?

If it's OK for you to work with command types, you can do the following.

You will have to pconfig("REPLACE",...) the following variable :

/AgentSetup/APPLICATION[.INSTANCE_SID].defaultAccount

and set this to the string "username/password", the agent will detect you are trying to set a username/password combination and it will encrypt the password for you (if the password was already encrypted, the agent will not encrypt it again).

Note that the INSTANCE_SID is optional, but will probably be necessary if your application can have multiple instances with different credentials.

Once this variable has been set, you will be able to use the %{username} and %{password} macro in any of your command type definitions for that application class.

In case you have to store multiple usernames and passwords on the same instance, you will have to use some “tricks”

Using defaultAccount in pconfig()

You will see that every pconfig() variable that ends on .defaultAccount will be encrypted when inserted in the pconfig database.

You can do the following

pconfig("REPLACE","/MYAPP/cred1.defaultAccount","user1/pass1");

pconfig("REPLACE","/MYAPP/cred2.defaultAccount","user2/pass2");

pconfig("REPLACE","/MYAPP/cred3.defaultAccount","user3/pass3");

if you want to use the first set of credentials, you can do the following

user_pass=get("/MYAPP/cred1.defaultAccount");

user=ntharg(user_pass,1,"/");

pass=ntharg(user_pass,2,"/");

Note that pass is still encrypted...

Now in the execute command type the following

execute("CMD_TYPE","ARGS","",user,pass);

This will allow you to switch between credentials.

Remember that any pconfig() access (even through get()) will go to filesystem. Therefore you might want to build a caching mechanism to avoid this. (this would be a bit too much for this topic I think)

Using encrypt()

Just like above, but instead of using the implicit encrypt ion like with defaultAccount... you just encrypt the password yourself (and store it as well)

I would use the encrypt() function

pass=encrypt("mypasswd","DES");

now you can use the execute function just like above

execute("CMD_TYPE","ARGS","",user,pass);

Why does security prompt only apply to OS commands?

PSL commands don't require/use a security context to run. They are all running in the PSL virtual machine and there are no users/privileges defined in the VM.

This means that every PSL process can execute any PSL instruction, without the need for validating credentials.

It would actually be strange to validate any credentials you would provide, since the virtual machine doesn't have any.

Can I control the number of OS processes that my KM spawns?

Sometimes when you develop a KM that does a lot of OS executions, the agent will start writing “Patrol could not fork a new process” in the Agent error log.

Obviously this is caused by the limit on the number of processes Patrol can spawn.

This can mean that the OS profile of the patrol user limits the max number of processes created. On UNIX this is usually a kernel parameter setting

This can be caused by excessive spawning of OS type commands, like multiple parameters in multiple instances spawning OS commands.

To see if this is really the problem, take a look at the /execsPerMin namespace variable by doing a

%PSL print(get(“/execsPerMin”));

In the system output window, or by opening the PAWorkRateExecsMin parameter in the PATROLAGENT.km

To prevent this from happening you can do some "tuning".

· Limit the number of OS processes. If you can, define a command type to execute the process instead of using system() or execute("OS",...) . Both methods would first spawn a shell interpreter that will then launch you command.
If you are executing an executable instead of a shell script, you should be able to define a command type that would prevent the spawning of useless shell interpreters.

· Limit the number of parallel OS executions... You can do that by adding some logic to each of the collectors that do shell execution

for example something like this: 

function nice_execute(cmdtype,cmd)

{

   local max_os_procs, num_procs, result;

   # If we are executing cmdtype OS, it counts for

   # at least to processes, if we execute another command type

   # we will presume it won’t create more than one process.

   num_procs=(cmdtype == “OS”) ? 2 : 1 ;

   # Get our max os processes

   max_os_procs=get("/max_OS_procs");

   # Sleep while our current OS processes is too high   

   while (get("/current_OS_procs") > max_os_procs)

   {

      sleep(1);

   }

   # Now add num_procs for this command to the current OS procs

   lock("os_procs");

   set("/current_OS_procs",get("/current_OS_procs")+num_procs);

   unlock("os_procs");

   # Do your OS stuff here

   result=execute(cmdtype,cmd);

   # And remove the processes again

   lock("os_procs");

   set("/current_OS_procs",get("/current_OS_procs")-num_procs);

   unlock("os_procs");

   # Return the result

   return(result);

}

The locking is necessary and should be done this way to avoid concurrency issues...

Of course this code will not work unless you set /max_os_procs to something.
And then again, this is just a gentlemen’s agreement.  Only executions that use the nice_execute() will be limited, someone else would still be able to execute system() behind our back.

· Probably the easiest way is probably to raise the maxprocesses limit in the kernel for the patrol user. Some UNIX variant will allow you to specify this per user. On other systems this will be a system wide setting.

Can I  prevent PSL from being executed based on credentials?

Yes, it is possible for you to store internal usernames and passwords, which would be useless to outside users. 

That means you can build your own user administration system that would only apply to PSL commands within the VM.

Just define a response function that will ask for username and password. Store this username in the configuration database, but store the encrypted password (not plaintext).

If you want to validate the user, create a response function that asks for username and password... encrypt the entered password and verify if this matches the encrypted password in the configuration file.

If you want to verify the supplied credentials with a specific set of OS credentials, you could execute a simple OS command with those credentials provided... if this succeeds, continue to execute the PSL, otherwise terminate.

If the OS credentials were wrong you will also get a runtime error on the system output window.

How do I call a C-function from PSL?

It's impossible to directly call a C function from within PSL.

The best way (least CPU) to get this done is write your own executable that can then be called by PATROL.

Even better would be if you would write an executable that would listen for commands (conversation type). That way you can create a channel to this executable and you would only have to start it once.

If you develop a good PSL library, the whole channel management (locking, restarting) can be hidden from the rest of the PSL code.

All PSL processes would be able to share the same channel using some primitive functions.

You might want to take a look at some database KM's since they all open a channel towards the SQL engine and share that channel.

I’ve heard about XPC, what is it about?

The XPC interface is NOT available as a separate toolkit to developers external to BMC. XPC is an interface only available to BMC internal developers since it has very specific linking requirements, and a strong dependency on the rest of the PATROL core code.

Actually the way XPC works is the same as what was described in “How do I call a C-function from PSL”.  It just starts a daemon (the XPC program) and the agent will open a channel towards this process.  All communication between Agent and XPC program is handled by sort of a channel-manager, but the same way as a proper PSL channel manager library would do it (locking, etc.)

There is nothing in XPC that can't be done with channel creation and "talking" to the executable yourself.

Again, XPC is not a toolkit.... and only available for internal developers, but it doesn't offer anything you can't do with plain PSL and C. 

Furthermore, the XPC functionality is available to everyone (even if you can’t use XPC as a library), by doing some proper c and PSL coding.

How does the agent determine an endless loop?

The are two configuration variable that determine when the agent considers a process to be in an infinite loop:

* pslInstructionMax

* pslInstructionPeriod

pslInstructionMax and pslInstructionPeriod work together.

If a PSL process executes more than pslInstructionMax PSL instructions within pslInstructionPeriod seconds, the PSL process incurs internal scheduling delay.

So both variables define the total PSL instructions a PSL process can execute without the delay.

The pslInstructionPeriod is a global timer for the agent. That means every [pslInstructionPeriod] seconds, all instructioncounters for all PSL processes that haven't reached pslInstructionMax yet, will be reset to zero.

However, if your process has reached pslInstructionMax already, it will not be reset, and the instructioncounter will just continue to increase for that PSL process.

When a process will be de-tuned, the agent will calculate the length of the delay in function of the total number of instructions, compared to the setting of Instruction Max.

A value of 0 for pslInstructionMax turns off the delay.

A value of 0 for pslInstructionPeriod makes pslInstructionMax

Default for pslInstructionMax: 500000

Default for pslInstructionPeriod: 7200

There is really nothing you can do to influence the value of the number of PSL instructions and there is no way to reset the instructionperiod counter (anyway, if it's too late and you are already on the "black list" changing this period won't help you).

What you can do, is set the pslInstructionMax counter to a higher value or even 0 :

* On the console using the %SET PSL_INSTRUCTION_MAX=[new number]

* In PSL using system("%SET PSL_INSTRUCTION_MAX=[new number]");

However, you might also want to take a look at your PSL program to see if it can't be changed, so it won't execute so much PSL functions...

Actually before your program starts executing, the agent will not know it actually contains an endless loop, but the agent will determine this at runtime, because the instruction count was to high over the period of time.

An instruction is any QUAD instruction. For example (ntharg, cat...) but also calls to user defined functions.  Actually it is counting the number of virtual machine instructions, "executed" by the VM instruction set processor.

To find out how many instructions are executed for a certain psl function, you can use the stand-alone psl compiler/interpreter.

If you do the following:

psl -O3P3 [your psl script]

Then the last line from the output will be:

[program name] [instruction count] quads, [total assigns] assigns

Should I save my code in the KM or an external PSL file?

When you save PSL code in the KM, the KM will contain all knowledge to manage a certain application. Only the KM file would be needed and no extra PSL scripts have to be shipped.

(That is if you don't require libraries as well)

The difference is in the KM file. Normally the KM will either contain the PSL script itself or a pointer to where the PSL script can be found.

If the PSL is part of the KM, the COMMAND_TEXT attribute for a parameter, menu command, infobox or (pre)discovery in a KM file would look something like this:

BASE_COMMAND = { 

{ COMPUTER_TYPE = "ALL_COMPUTERS", COMMAND_TYPE = "PSL",

COMMAND_TEXT = 852329761 "phome = get(\"/patrolHome\");\

hname = get(\"/hostname\");\

portno = get(\"/udpPort\");\

mach = get(\"/appType\");\

[snip]

set(\"value\",int(siz));"}

Otherwise it would contain the LOAD statement like shown here:

BASE_COMMAND = { 

{ COMPUTER_TYPE = "ALL_COMPUTERS", COMMAND_TYPE = "PSL",

COMMAND_TEXT = LOAD "usr_proc_collector.psl"}

},

When you want to write a patch for a certain KM, and the PSL is not inside the KM file, you can just replace the exisiting psl file with a newer one. Any other PSL files that haven't changed would not be affected. Also, the file sizes for each of the files will be very small and in some cases that is a good thing when doing small upgrades/changes.

But the customer would need all files to be there before a KM would be functional, more files can mean more administration on his side... or more work if he wants to repackage your KM.

Another benefit of having it in a separate psl file is if you are using CVS or similar source control mechanism.  The granularity for checking in files will be smaller and it’s easier to track changes on specific files.

Besides that, when PSL is stored in an external file, you can just open it with a text editor and copy/paste or even read the code without having to deal with escaped quotes, newlines and such.

If all PSL is part of a KM, and you want to bring out a newer version of your KM, you only have to ship one file, no matter how much you changed. However, by shipping a new KM file, you might overwrite changes made to any other script inside the KM. I mean if the KM contains all PSL and if someone wrote additional parameters in the KM, or changed some code in one of the parameters, you will automatically overwrite that if you replace the old KM with a new KM file.

The benefit of storing everything inside the KM is that you don't have to determine dependencies and find out which files have to be shipped as well. Only the KM file will do.

How does the ExtraFilesList work?

The ExtraFilesList is a console enhancement to the Commit KM feature that is aimed to allow committing of file types that are not normally handled by Commit. The normal Commit only sends files with suffixes ".km", ".psl", and (sometimes) ".bin". It is not normally sending files with suffixes ".lib" or any other special scripts that should be part of the package. 

· Create a new standard parameter called "ExtraFilesList" in the desired KM. Use this exact spelling and capitalization of letters.  The command type of the parameter and scheduling information is ignored so do not change it. 

· Set the parameters "active toggle" so that the parameter is not active.  This will make sure it doesn't run and get compile errors on the agent. 

· For the newly defined parameter a list of files can be added to the command text window. Each file in this list will be committed to the agents when the KM is committed. A distinction is made between PSL library files and any other file. The version of PSL libraries is checked before the library is actually sent to a particular agent. If the version of the library on the agent is the same as the library that is to be committed it is not sent (since they are the same). All other files are sent unconditionally. Each file should be specified on a separate line and the keyword LIB or EXTRA should prefix the filename. 

Some examples :

EXTRA "SomeFile" 

EXTRA "../SomeOtherFile" 

LIB "SomeFile.lib" 

EXTRA "adduser.sh" 

This assumes that the file is in either the global/local PSL directory. If it is not the path to it must be specified relative to the local/global PSL directory. The agent always places the file relative to the global PSL directory. 

The first line indicates that the file SomeLib.lib should be sent to the agent during a commit. 

The LIB keyword means that the agent will check the version of the library before the library is transferred to the agent. If the library versions differ or the agent does not have a local copy the library is sent to the agent.

The second line also specifies that a PSL library should be sent to the agent, but this time the EXTRA keyword is used. This means that no version checking is done and the transfer occurs unconditionally. 

The last two lines transfer images to the agent (just for fun). These could be any files at all, and the file doesn’t even have to be a PATROL file. Note that the path is relative (i.e. it contains a…), it is relative to either the local psl directory ($HOME/patrol/psl) or the global psl directory ($PATROL_HOME/lib/psl). 

When the console attempts to find the specified file is always considers the local directory first. The files will end up on the agent in the same location relative to the global psl directory, regardless of whether the file was taken from the local directory or the global directory.

Also notice that each line is preceded by a # character, this is because the # character is used as the comment character by most shells and PSL, thus if something does accidentally activate the ExtraFilesList parameter nothing will execute since every line is a comment. 

Can I programmatically create recovery actions?

Some people would like to know if there is a way to set recovery actions (just running predefined OS scripts) using PSL, in order to be able to manipulate the recovery actions. (Also no console can/should be used).

This is not really that easy with the way recovery actions are defined and executed.  

Let’s say you know exactly which parameters you would like to execute recovery actions for (and you are able/willing to change the KM’s).

What you could do is write a generic recovery action that would determine which psl script you should execute (you can call the script using PslExecute – take a look at a PslExecute question in this FAQ).

In this generic recovery action you can create a loop with all escalation routines you want.  The loop termination condition could be to check on “status” to see if the parameter is still in the same alarm range.

If you want to take it further and don’t even want to add a generic recovery action, then you could write your own notification action on the standard event 11. Event 11 will be triggered if the value falls within the defined alarm ranges. 

Whenever this event is triggered, the notification action will execute.

This will allow you to run at least 1 recovery action per alarm range, without the need for a console (since it can all be code).

The only problem is that there will only be one single event per parameter as long as the values keep falling within the same range. 

I mean: parameter X goes into alarm... the stdevent will be triggered and you notification action will fire. This notification action has to take care of the recovery action. If the parameter is still in the same alarm range at a next polling interval, no additional events will be fired. (Your parameter will still be in alarm)

Of course you can also write a full escalation system yourself, because you know that if parameter X is back in OK state, standard event 9 will be triggered, so as long as you didn't receive event 9 for this parameter, you should still be processing recovery actions.

All this could be written in a standalone KM if you want. (Of course besides the changes to the standard event catalog)... That way an operator console can select recovery actions. You can define a "recovery action interval" or escalation routines and the commands to execute... You could also report on when a certain recovery action was run, how long it took to complete...

Definitely, there is a lot you _can_ do.

How does the /AgentSetup/[APPL].filterType  variable work?

This variable works in tandem with the /AgentSetup/[APPL].filterList configuration variable, which can be changed using (x/w)pconfig.

Let's presume you have an application and its discovery contains this code:

for (i=0; i< 10 ; i++)

{

  name="inst_".int(i);

  if (create(name,name,OK))

  {

    print("Instance : ".name." create succeeded\n");

  }

  else

  {

    print("Instance : ".name." create failed\n");

  }

}

This will create 10 instances: inst_0 ... inst_9

Now, if you set

/AgentSetup/[APPL].filterList = "inst_1, inst_3, inst_5"

/AgentSetup/[APPL].filterType = "exclude"

Then these three instances will not be created, but all the rest will.

if you set

/AgentSetup/[APPL].filterList = "inst_1, inst_3, inst_5"

/AgentSetup/[APPL].filterType = "include"

Then only these three instances will be created, all the rest will be suppressed.

How can I use Command Types in (Pre-)Discovery?

A command type defined on application level (and not on computer class level) has to run in context of an instance of this application.

This instance should be specified in the 3rd argument of the execute function, in case you are using the execute function with such a command type in a discovery script.

When execute is called in an infobox, menu command, parameter or recovery action, PATROL will find the nearest ancestor instance himself. However, since discovery is ran on application level, there is no way the agent can determine the instance you would like to use...

In case you want to run execute() after you have already created an instance, you can specify that instance.

For example: discovery

...

create(inst,inst,OK);

...

execute(cmdtype,cmd,inst);

...

will work.

In case you are trying to discover which instances you should create for your application class (for example CHILD.km), you can define the command type in another KM that can even possibly be the parent of CHILD.km

(for example PARENT.km)

In this case, PARENT.km should already have at least one instance ... for example; PARENT/TOPLEVEL could be the unconditionally created parent instance.

Now in discovery of CHILD.km, you can call execute() like this : 

execute("CMDTYPE","cmd","/PARENT/TOPLEVEL");

Another option (not really advised) would be to create an invisible instance of your application (unconditionally).

In your discovery you can do the following to create an invisible instance.  Note that the name “invisible” is not important, but the fact that this create statement only has one argument is !

if (!exists("invisible"))

{

# Create invisible object

create("invisible");

}

Now every execute() can use "invisible" as the instance to provide the context.

execute(cmdtype,cmd,"invisible");

How do I easily extract a filename from a path?

The following piece of code will do it for you

# Break up pathname, both backslash and forward slash are allowed

pieces = ntharg(path, "1-","\\/","\n");

# Get the last piece

filename = tail (pieces,1);

Is there an isalpha() equivalent in PSL ?

This function doesn't really exist in PSL, but the example below will do the same as isalpha()

function isalpha(text)

{

  return(!trim(toupper(text),"ABCDEFGHIJKLMNOPQRSTUVWXYZ"));

}

text="AbG#$D";

print("ISAPLHA ".text." = ".isalpha(text)."\n");

text="AbGD";

print("ISAPLHA ".text." = ".isalpha(text)."\n");

text="%AbGD";

print("ISAPLHA ".text." = ".isalpha(text)."\n");

text="hElLo123";

print("ISAPLHA ".text." = ".isalpha(text)."\n");

text="_bla1";

print("ISAPLHA ".text." = ".isalpha(text)."\n");

Just run it in a PSL task window to see how it behaves.

Just remember you might want to add 0-9 and characters like _ to the list as well.  For sake of convenience I renamed the function to isvalid because you could add any character you want to the list of valid characters.

function isvalid(text)

{

  return(!trim(toupper(text),"ABCDEFGHIJKLMNOPQRSTUVWXYZ0123456789_"));

}

text="AbG#$D";

print("ISVALID ".text." = ".isvalid(text)."\n");

text="AbGD";

print("ISVALID ".text." = ".isvalid(text)."\n");

text="%AbGD";

print("ISVALID ".text." = ".isvalid(text)."\n");

text="hElLo123";

print("ISVALID ".text." = ".isvalid(text)."\n");

text="_bla1";

print("ISVALID ".text." = ".isvalid(text)."\n");

How do I read I file with garbage (and NULL strings ‘\0’) ?

If you want to read a file with garbage in it (Even \0 characters), read() and cat() functions will fail each time a \0 character is encountered. The file is just closed and the rest of the file is never read.

The workaround below should only be used it in case your file contains \0 characters because it will be a slower than reading a file the normal way.

# Enter your filename here

fullPath="....";

fp = fopen(fullPath, "rb");

while (1)

{

  input = read(fp,1);

  pos=ftell(fp);

  contents=contents.input;

  if (pos==oldpos) { last;}

  oldpos=pos;

}

close(fp);

print("CONTENTS = ".contents);

How do I copy a binary file in PATROL?

Actually this problem is related to the problem described before (“How do I read I file with garbage?”).

If you try to copy a binary file, you will see that the file will not be copied completely because each time PATROL reads a \0, PATROL will use this as a string terminator.

When you read a file, you put the result in a variable. Although the variable will contain all the data, you won't be able to get to the data, because when you hit a \0 character, it is considered to be the end of a string.

Below you will see the best attempt in writing a binary file copy script. This copy will behave correctly even with null characters in the middle of the file, unless the binary file itself contains null characters at the end (can’t fseek beyond EOF).

You will see that the method for reading the input file is different than the method used in the previous question (this one uses block mode reads). This method will be significantly faster if you have only a couple of \0 characters in the file.

Comments to the source were added, so it would be easier to understand what’s happening.

infile="c:\\temp\\in.bin";

outfile="c:\\temp\\out.bin";

#infile="/tmp/in.bin";

#outfile="/tmp/out.bin";

fpin=fopen(infile, "rb");

if (!fpin)

{

  print("Input file : ".infile." could not be opened !\n");

  exit;

}

fpout=fopen(outfile, "wb");

if (!fpout)

{

  print("Output file : ".outfile." could not be opened !\n");

  exit;

}

# Get file stats

size=ntharg(file(infile,1),5);

while (1)

{

  # Save the pos before reading

  pos_before_read=ftell(fpin);

  # Read till EOF...(actually we would like to read till first \0

  input = read(fpin,-1);

  # Get the actual "visible" length

  len=length(input);

  # Rewind input pointer for reading.

  # Put it one location after the length of the string

  # we just read

  pos=pos_before_read+len+1;

  fseek(fpin,pos,0);

  # Write only the "visible" input in our output file

  write(fpout,substr(input,1,len));

  # This will move the write pointer one position further and

  # move it behind a NULL character, thus adding a \0...

  fseek(fpout,1,1);

  # Check if we really hit the EOF

  if (pos > size) 

  { 

    last;

  }

}

close(fpin);

close(fpout);

How do I clear text parameters?

The PATROL agent will not append to text parameter unless you explicitly ask it to.

If you just set the text parameter with a new value each time, you won't have a problem.

However, if you keep the parameter open on the console (especially NT console), it will seem like it's appending, although this is not the case (just a visual gotcha).

There is a namespacevariable "/clearText" which will allow you to clear the console output window.  I've seen that the UNIX console interprets this correctly, but an NT console might not clear the window...

That means instead of setting like this

set (base."/SampleText/value",variable);

You should set like this

set (base."/SampleText/value",get("/clearText").variable);

But again, this is just a visual thing that will force the console to clear the window.

You should be careful when you do things like this:

olddata=get(base."/SampleText/value");

set(base."/SampleText/value",olddata.newdata);

Because this will append data and your parameter will start growing. This is NOT is history, because a text parameter doesn't write history, but run-time memory (the agent has to store the data somewhere).

If you are not careful your agent might "explode" ... (see “How do I append text to a text parameter?”)

What is so special about “/clearText”?

The text parameter and system output window are a kind of a UNIX terminal window, but only recognize a limited number of escape sequences.

Actually the only escape sequences that will be properly handled are “clearscreen” character sequences for ANSI and VT100 terminals (Esc [H Esc 2J and Esc [;H Esc 2J). Sending this sequence will clear the window.

Sequences to change the color, appearance, or any other VT100 terminal sequences will not be recognized.

Actually when you do a 

print(get("/clearText"));

you will print the ANSI terminal clearscreen escape sequence and this is recognized by the text output window.

Just to show you, type the following

%PSL printf("%c[H%c[2J",27,27);

And you will see the screen will clear as well.

How do I append text to a text parameter?

First of all some people think that text is automagically appended to a text parameter by PATROL, because when they open a text parameter, they will still see the previous text in there.  Actually this is the console appending it for you (not the agent), and this is just a visual effect.

This is more done for the convenience of the operator, to be able to see the history for as long as he has the window open.

If you want to append the text in the agent (so it will also work if the text window is not open), you will have to add the text to the parameters value.

Just like any other parameter, the text parameter will show the current value.

If you want to set the parameter with a certain string, you can do it like you would do it with any other parameter:

set("value","mytext");

If you want to append text to the current text, you can do the following

oldtext=get("value");

newtext=oldtext."Some new text";

set("value",newtext);

However, when doing this, you have to be careful ! The value of the parameter is stored inside the agent’s memory. If you keep appending to the value of a text parameter, this can grow very big and the agent will just continue to grow. This behavior could be seen as a "memory leak", because you will only add data to a parameter, but never clean it up!

Therefore you shold limit the amount of text that is saved in the parameter.

An easy way to do this is:

oldtext=get("value");

newtext=oldtext."Some new text";

# Only store last 100 lines

newtext=tail(newtext,100);

set("value",newtext);

The get("/clearText") will not clear the allocated memory inside the agent.  I will just tell the console to clear the text output window.

How do I synchronize my channel?

Let’s fist explain the problem with channel synchronization …

For example: you are using popen(), read() and write() to simulate an interactive telnet session and have a function (below) which submits a command to the telnet session and returns the response.

Once the channel is open, you call this function several times in succession.

function mcomm(channel,command)

{

write(channel,command."\n");

output=read(channel);

return(output);

}

a=mcomm(handle,"echo apple");

b=mcomm(handle,"echo banana");

c=mcomm(handle,"echo pear");

Normally you would expect that

a = apple

b = banana

c = pear

What you might see is :

a = 

b = apple

banana

c = pear

or

a = 

b = 

c = apple

banana

pear

Definitely this is an indication about a synchronization problem in the channel communication.

What you want to do is force read() to wait but also allow for the case where the submitted command may not return any output.

As with any component that communicates with another component, you have to define some type of protocol.

By adding a minor convention to your code you can make the communication more stable. The "trick" is to wait until you get a termination signal... (In this case a very bizarre piece of text). Of course the end-of-output could also be signaled by a prompt or something else.

If the sender can tell you how many bytes it will send, you could determine the end-of-output by counting the number of returned bytes.

One of the possible ways of getting it done is shown below:

function mcomm(channel,command)

{

  local data,output;

  output="";

  term="@@@_THE_END_@@@";

  term_cmd="echo\necho ".term."\n";

  write(channel,command."\n".term_cmd);

  while (data=read(channel))

  {

    output=output.data;

    if (grep("^".term."$",output))

    {

      output=replace(output,term."\n","");

      last;

    }

  }

  return(output);

}

Depending on the product you are talking to, there might be other and better solutions.

How is the clicker supposed to behave?

A lot of people have told me about problems they have with the clicker element in the response function.  Most of the times because of the inconsistency between the UNIX and the NT console, but sometimes also because they didn’t clearly understand how it is supposed to work or behave.

Before I get into the known problems, I would like to explain how the clicker is supposed to work.

You know you have to specify a min value and a max value and a default value.

Let's refer to them as min,max and default.

A valid clicker specification would be

[R_CLICKER,"MyClick",min,max,default]

· If default doesn't lie between min and max, PATROL will take for default a value of (min+max)/2 !

· If min > max then min=0 and max=100

On the UNIX console there is/was indeed a problem.

The clicker window is sized according to the size of the default value.

The value will be correct (you just have to scroll in the small window containing the number to see the actual value)

A possible workaround is to make sure the default value contains at least as much characters a there are character in min and max. (A leading zero will not work for default)

For example:

min=-99;

max=10;

def=-50;

if (def>max) { def=max; }

if (def<min) { def=min; }

lenmin=length(int(min));

lenmax=length(int(max));

lendef=length(int(def));

# Trap clicker problem...set default to something else

if ((lendef<lenmin) || (lendef < lenmax))

{

  def=(length(min) > length(max)) ? min : max;

}

printf("min=%i - max=%i - def=%i\n",min,max,def);

Of course, this limits the use of this control a lot, because you will not be able to set default really as you would like to. (But in cause you don't mind, this might be useful).

An alternative might be to use the slider.

Why does pconfig(“GET”…) always return an extra newline?

The pconfig("GET") function returns the default and current values of variable in a PSL list with the following format:

default_value\n

current_value

The default value gets picked up from the configuration specified in the config.default file. If the variable is not specified there (or is empty), you will just see the newline...

Instead of using pconfig("GET"), you might want to use

pconfig("FORMAT",32|2,"","","/PCFGVAR");

This will return "just" the current value.

How do I set an environment variable on instance level ?

In case you want to launch OS commands from parameters in a specific instance, it might sometimes be useful to set the environment that the OS processes will inherit.

The environment attribute can contain a comma-separated list of environment variables.

For example :

set("/APPL/INST/environment","ENV1=X,ENV2=Y");

How do I only read new data from a log file?

Some people try to write code where they read from a log file in a parameter - but each time the parameter runs, they only want to read new that has been added to the file, and not read the whole file again.

Therefore I wrote a piece of code that shows how you can do it.

Of course, this code does not handle files that may be replaced/deleted... but it will help you get started.  (If you want to add that logic, you might want to check if the creation time of the file changed).

filename="c:\\temp\\testfile";

lastpos=get("lastpos");

print("The last position was : ".lastpos."\n");

# Open the file

chan=fopen(filename,"r");

print("The channel is : ".chan."\n");

# Reposition the filepointer

if (fseek(chan,lastpos,0)==0)

{

print("Reposition successfull\n");

}

else

{

print("Reposition failed\n");

}

# Read till EOF

data=read(chan,-1);

# Save the filepointer

lastpos=ftell(chan);

set("lastpos",lastpos);

print("The new position is : ".lastpos."\n");

# Print the data

print("The data was :\n#".data."#\n");

# Close the file

close(chan);

How do I start a daemon/background process from PATROL?

To start a daemon process from PATROL, use

chan=popen("OS","executable to start");

close(chan);

The close() will close the channel without actually stopping the external process...

If you would do close(chan,3); the external process would be killed as well.

How do I make the system() function non-blocking?

The system function itself if blocking and cannot be made non-blocking (yet).  However there are other ways to achieve the same solution.

This non-blocking behavior is highly requested when the executable that is called can sometimes hang and therefore block the KM.

This is indeed a very complex issue. The reason why it is so complex is because the execute and system commands don't have a timeout argument, so we will have to provide it somehow...

Instead of just telling you how to solve the problem "theoretically" I have a script that illustrates how you can solve the problem.

############################################################

# Author : Geert De Peuter

# BMC Software Developer Connection

############################################################

DEBUG=0;

############################################################

# This function will execute an OS command with timeout.

# Arguments

#  cmd : The OS command you want to execute

#  timeout : the timeout

############################################################

# This global variable will contain the errors that happened

# if there are any.  The variable is exported to avoid

# compiler warnings.

export em_error;

function execute_monitored(cmd,timeout)

{

  local chanid, channame, data, output, psl, chan, myerr;

  # This should be a global variable that will return the

  # error status

  execute_monitored_error="";

  if (DEBUG)

  {

    print("execute_monitored started with ".

          "args(".cmd.",".timeout.")\n");

  }

  # This is the name of the shared channel we will use

  # We will need a unique channel for each execution, 

  # therefore this routine

  lock("CHANIDGENERATOR");

  chanid=get("/CHANIDGENERATOR");

  chanid++;

  set("/CHANIDGENERATOR",chanid);

  unlock("CHANIDGENERATOR");

  # Now make a channelname

  channame="CHANMON_".chanid;

  if (DEBUG)

  {

    print("create a channel to ".cmd."\n");

  }

  chan=popen("OS",cmd);

  if (!chan_exists(chan))

  {

    em_error="ERROR : Channel could not be opened!";

    return("");

  }

  if (DEBUG)

  {

    print("Channel is now ".chan."\n");

  }

  # We have to share the channel, because we want 

  # another PSL process to be able to kill it

  if (share(chan,channame) == -1)

  {

    if (DEBUG)

    {

      print("ERROR : Share channel failed\n!");

      print("        Trying to recover\n");

    }

    # Forcefully close channel if already exists

    close(channame,3);

    # Try to share the channel again

    if (share(chan,channame) == -1)

    {

      em_error="ERROR : Impossible to share the channel!";

      return("");

    }    

  }

  if (DEBUG)

  {

    print("Channel is shared as ".channame."\n".

          "Channel(id=".chan.") was opened and shared\n".

          "Will start reading now, this might block the ".                    

          "channel.\n".

          "The channel will be closed by the channel monitor ".

          "in ".timeout." seconds\n");

  }

  # Start a timeout monitor to kill any overdue processes.

  psl = "sleep(".timeout.");\n".

         # Check to see that the channel still exists.  If not,

         # the process completed and closed the channel.

        "if (chan_exists(\"".channame."\"))\n".

        "{".

           # If we're here, and the channel still exists, then

           # the channel is still blocked - close it.

        "  close(\"".channame."\", 3);".

        "}";

         # Channel closed by itself

  # Launch the monitor

  PslExecute("CHANMON_".channame, psl);

  # This will block the channel...

  while (1)

  {

    data=read(channame,0);

    myerr=errno;

    if (myerr==91)  #E_PSL_UNBLOCKED_BY_CLOSE

    {

      if (DEBUG)

      {

        print("Unblocked by close\n");

      }

      last;

    }

    elsif (data=="")

    {

      if (DEBUG)

      {

        print("Read returned with errno ".myerr."\n".

              "Possibly the command was terminated\n");

      }

      last;

    }

    # Save the data

    output=output.data;

  }

  if (DEBUG)

  {

    print("Channel unblocked ...\n");

  }

  # Cleanup

  # Destroy the channel if still opened

  if (chan_exists(channame))

  {

    close(channame,3);

  }

  if (DEBUG)

  {

    print ("End of execute_monitored\n");

  }

  return(output);

}

print("Output of 'dir', timeout 10 seconds :");

print(execute_monitored("dir","10"));

print("\n\n\n");

print("Output of 'ls', timeout 10 seconds :");

print(execute_monitored("ls","10"));

print("\n\n\n");

print("Output of 'sleep 20', timeout 10 seconds :");

print(execute_monitored("sleep 20","10"));

print("\n\n\n");

print("Output of 'PatrolCli' (blocks), timeout 10 seconds :");

print(execute_monitored("PatrolCli","10"));

print("\n\n\n");

Instead of using system() or execute() we should use to start the process using popen() … see chan=popen("OS",cmd). This will create a channel to the process you are executing and therefore give us some control.

The problem we still have with popen() is that a read on a channel that returns nothing will still block (and a hanging process will block as well).

So, whenever you want to execute a dangerous task, you have to start another job at the same time that can kill the task if the execution is taking too long. (Or possibly blocked in a read).

When I say killing a process can be done by simply closing the channel with 3 as the second argument, but if we want another PSL process to be able to close the channel, the channel should first be shared.

So what we have already as requirements are

- Command has to be executed using popen()

- Channel from popen() has to be shared so another PSL process can close it

- Another PSL process has to be started so it can monitor if the channel times out.

If we know this code will be executed inside a parameter, we might use one of the parameters in the instance to actually monitor the opened channels for timeout (this parameter will be a channel monitor).

Because sometimes the code has to be executed from discovery, we had to find another way to create an additional (detached) PSL process.

The function that should be called is “execute_monitored()” that will do everything needed to safely execute a command with a certain timeout.

You have to be careful not to use this execute_monitored() for every system function you want to execute because there is an extra overhead.  However, in case you really need this functionality, this is probably one of the easiest ways to do it.

How do I get all parameters in the namespace?

Here is one of the ways to do it...

function print_parameters(instpath)

{

  local parm, parms, parmpath;

  parms=get_vars(instpath,"subnodes");

  foreach parm (parms)

  {

    parmpath=instpath."/".parm;

    if (get(parmpath."/__type__") == "PARAM_INSTANCE")

    {

      print(" PARAMETER : ".parmpath."\n");

    }

  }

}

function print_instances(applpath)

{

  local inst, instances, instpath;

  instances=get(applpath."/instances");

  foreach inst (instances)

  {

    instpath=applpath."/".inst;

    print(" INSTANCE : ".inst."\n");

    print_parameters(instpath);

  }

}

function print_applications()

{

  local appl, applications, applpath;

  applications=get_vars("/","subnodes");

  foreach appl (applications)

  {

    applpath="/".appl;

    if (get(applpath."/__type__") == "APPLICATION")

    {

      print("APPLICATION : ".appl."\n");

      print_instances(applpath);

    }

  }

}

print_applications();

What is the __type__ attribute used for ?

__type__ is an object attribute that describes where you are in the namespace. It is frequently used when walking the PATROL namespace.

It can contain values like:

- COMPUTER

- APPLICATION

- APP_INSTANCE

- PARAM_INSTANCE

pconfig "directories/nodes" usually also define __type__ as DATA_REPOSITORY

How many significant digits does PATROL use ?

Internally all values are stored as an ascii representation of a double.

Whenever mathematical operations are performed, the value will be converted to an integer (usually 4 bytes) or a double (usually 8 bytes) and then converted back to a string. The type used for doing a mathematical operation really depends on the operation.

For example +, -, \, * all work with double precision. A << or >> (bit shifting) will work with integer precision (since they are bitwise operators).

What is the maximum length of a string in PATROL?

The maximum string variable value is limited to the memory you have in your box, or the memory the PATROL process is allowed to use (if your OS allows you to specify this).

How do I change directory (cd) in PSL?

One of the reasons why you would like to do this, is if you want to run a program that expects to be executed from a certain location.

The way to do this on NT is quite different than on UNIX.  The code below will show you one of the ways to lauch a command from a certain location.

This example will launch Kmtool …

OS=get("/appType");

phome=get("patrolHome");

if ( OS == "NT")

{

  drive=ntharg(phome,"1",":");

  if (drive)

  {

    command=drive.": & cd \"".phome.

            "\\lib\\knowledge\" & kmtool ";

  }

  else

  {

    command="cd \"".phone."\\lib\\knowledge\" & kmtool ";

  }

}

else

{

  command="cd ".phome."/lib/knowledge && kmtool ";

}

kmtooloptions="-report -set CPU.km FILESYSTEM.km";

print(system(command.kmtooloptions));

How do I create a simple channel to Perl for exhaustive grep ?

A comment PATROL often receives is that it is not very good at string manipulations.  One of the ways to overcome this problem is passing this intensive job of to for example a PERL interpreter.

The code below will show you how you can do that…

These two programs work in tandem and talk over a channel.

PSL Program
function chan_grep(channel, pattern, text)

{

  local filter, eot, len, data;

  # Used by the perl script

  filter="\n@#!FILTER!#@";

  eot="\n@#!EOT!#@\n";

  if (chan_exists(channel))

  {

    # Make sure noone will use the channel while we are

    lock("chan_grep_".channel);  

    # Write the filter definition, the text and the EOT

    # message.  It would be possible to have multiple

    # filter conditions

    # But you will need to learn regular expressions.

    write(channel,filter.pattern."\n".text.eot);

    # After EOT, perl script will return the length of the output

    len=trim(readln(channel),"\r\n");

    # If there is more than 0 to read, get it...

    if (len>0)

    {

      data=read(channel,len);

    }

    # Free the lock

    unlock("chan_grep_".channel);

  }

  return(data);

}

chan=popen("OS","perl /tmp/xxx.pl");

text="1: aaa bbb ccc\n2: aaa xxx ccc\n3: aaa bbb ccc\n";

output=chan_grep(chan,"xxx",text);

print("\nOutput 1 =".output);

output=chan_grep(chan,"bbb",text);

print("\nOutput 2 =".output);

output=chan_grep(chan,"xyz",text);

print("\nOutput 3 =".output);

close(chan);

Perl Program

$|=1;

while (<STDIN>)

{

  if (/^@#!FILTER!#@\s*(.*)/)

  {

    $filter=$1;

  }

  elsif (/$filter/)

  {

    $output.=$_;

  }

  elsif (/^@#!EOT!#@/)

  {

    print length($output)."\n".$output;

    $output="";

  }

}

Why would I use %ECHO instead of print() in infoboxes ?

Quite often you will see that infoboxes define the command type as OS and then execute the command:

%ECHO %{myvar}

instead of using a PSL command type and writing

print(get(“myvar”));

There is a reason why you will see this.  For every PSL process, the agent will create a runtime object (rtcell) and then schedule that for execution.  This can be a little bit time consuming in case you are waiting for the command to come back (and in case of an infobox, you are probably waiting for more commands to be executed).

The special behavior of the % commands (like %DUMP, %PSLPS, %ECHO, …) is that they are not PSL commands and will be “recognized” whenever you try to do a system execution (like in the system output window).

This means %ECHO can be used as a very lightweight method to get simple namespace data while bypassing the PATROL runq and PSL scheduler.

What is the performance impact of =~, !~, grep() and index() ?

The optimizer will not change one command with another, each function has it's own reason of existence.

I'll try to spend some time on each of the different functions and hope this helps you understand why it is working this way.

text =~ pattern

Returns 1 if regular expression pattern matches the text, otherwise zero. The text can be multi lined.

This is an operator. An operator takes two arguments are returns a third argument in a single QUAD instruction. The operation is a regular expression match and it will unconditionally behave like a very fast version of grep()... no options, no checking, no return values, besides 1 or zero.

Whenever the optimizer can, it will "pre-execute" this operator at compile time. That means if the two arguments are static (hardcoded), using =~ will have no runtime overhead.

The execution of =~ is fast (much faster than grep and I guess a little bit faster than index), but

* Will return 0 or 1

* Will always use regexp (be careful when your pattern contains regexp characters)

grep(pattern,text,options);

Returns the lines of text that match pattern.

This is a build-in function and will therefore require a quad more to execute than an operator.

Has options to tweak the execution, and will try to determine if pattern contains regexp or not. It works line by line and if pattern does not contain regexp it will use a sort of index() function (without compiling and executing the regular expression) to optimize for speed.

This is the slowest of them all on a line mode, but a lot faster then writing your own using =~ and foreach line on a block of text.

The optimizer will not "pre-execute" the function, even if the arguments are hardcoded (or static).

index(text,string);

Does not work with regular expressions. It will return the position of string within text. Works on multilines and very useful to see if a certain word appears in a block of text.

This is a PSL function with less overhead than grep().

The optimizer will "pre-execute" in case both arguments are hardcoded (or static)

Can I use grep to look for multiple strings?

Sure, the PATROL grep() function supports multiple string matching.

Try this :

grep(“string1\\|string2”,text);

I found a limitation in PSL readln() …

Indeed, the limitation of the readln() is 4095 characters + 1 for the null byte.

Below is a function call SafeReadln which will readln and if the err 57 occurs, try reading again until the EOL is seen.

longfile = get("/patrolHome")."/longfile.txt";

function Safe_Readln(chan)

{


local psl, x, data;

pos = ftell(chan);

x = PslDebug;

PslDebug = 0; # Turn off runtime error reporting

data = readln(chan);

while (errno == 57) # E_PSL_READLN_TRUNCATED

{

pos = pos + 4095;

fseek(chan, pos, 1);

data = data.readln(chan);

}

PslDebug = x; # restore runtime error reporting

return data;

}

readchan = fopen(longfile, "r");

if (!chan_exists(readchan))

{

print("error <".errno."> opening <".longfile.">\n");

exit;

}

longbuf = Safe_Readln(readchan);

print("========== longbuf =============\n");

print("long buf is <".length(longbuf)."> bytes long\n");

print(longbuf);

print("=========- end of longbuf ========\n");

How do I smartly execute PSL scripts from within PSL?

A basic way to launch a PSL script from within PSL is something like this:

pslFileName = "someOtherFile.psl";

pslScript = cat(pslFileName);

if (pslScript)

{

PslExecute("Test", pslScript);

}

This method has the advantage that the parent script does not wait for the child script to complete before continuing.  However, the child script could become quite large and you could expect performance problems using this method.

Another disadvantage is that the PSL script will be compiled for every execution.

In case you really need this functionality, you could put as much code as possible in a library to minimize that overhead.

If you don't want to use libraries, you could use compiled PSL to limit the overhead. If you call the psl standalone compiler with a -b option on a psl file, it will save the compiled psl. 

You could then call the PslExecute with the content of this file.

You could then create a function that wraps the execution in a proper way with minimal overhead then you should try to understand the code below.

It will cache the compiled PSL and recompile PSL if the external PSL file has changed.

function CachePslExecute(script)

{

  scriptpath=get("/patrolHome")."/lib/psl/".script;

  pslscript=scriptpath.".psl";

  # Increment the counter

  counter=get("/__counter");

  counter++;

  set("/__counter",counter);

  timestamp=file(pslscript);

  timestampvar="__timestamp_".script;

  cachevar="__cache_".script;

  if (get(timestampvar) != timestamp)

  {

    if (timestamp)

    {

       # Save the timestamp

       set(timestampvar,timestamp);

       # Compile psl

       system("psl -b ".pslscript);

       # Save the result in the cache

       binscript=scriptpath.".bin";

       if (file(binscript))

       {

          set(cachevar,cat(binscript));

          # Remove the binary file ... no need to keep it

          remove(binscript);

       }

       else

       {

          print("Binary file was not generated !\n");

       }

    }

    else

    {

       print("File was removed ... “.

             “will continue to use cache !\n");

    }

  }

  binpsl=get(cachevar);

  if (binpsl)

  {  

     PslExecute(script.counter,binpsl);

  }

  else

  {

     print("No binary PSL was found !!\n");

  }

}

# Will execute and cache the script %PATROL_HOME/lib/psl/tester.psl

CachePslExecute("tester");

Can I use hashes in PSL?

Unfortunately hashes are not a native datatype of PSL.  Of course, you could always create a simple hash function library.

##########################################

# HASH FUNCTIONS

# --------------

# USED DELIMITERS

# \A \B     : Hash

# \N        : Replacement for \n

#

# A hash is like a two element array,

# each element is stored as a key,value pair like this

#     \Akey\A\Bvalue line 1\Nvalue line2\B\n

# hash keys are unique for a hash

# The hash value can be a newline separated list.

##########################################

function hash_dump(hash)

{

  hash=replace(hash,"\A\B"," => ");

  hash=replace(hash,"\A","");

  hash=replace(hash,"\B","");

  hash=replace(hash,"\N","\n");

  return(hash);

}

function hash_keys(hash)

{

  local keys;

  # Get the keys from a hash

  keys=ntharg(hash,"1","\A","\n");

  return(keys);

}

function hash_iskey(hash,key)

{

  local iskey;

  # See if key is already defined in the hash

  iskey=index(hash,"\A".key."\A") ? 1 : 0;

  return(iskey);

}

function hash_set(hash,key,value)

{

  local matchline;

  value=replace(value,"\n","\N");

  # Add key value pair to hash... 

  # if key already exists in hash them replace value

  if (hash_iskey(hash,key))

  {

    # replace, this is a key

    matchline=trim(grep("\A".key."\A",hash,"tb"),"\r\n");

    hash=replace(hash,matchline,"\A".key."\A\B".value."\B");

  }

  else

  {

    # Just add it

    hash=[hash,"\A".key."\A\B".value."\B"];

  }

  return(hash);

}

function hash_get(hash,key)

{

  local matchline, value;

  # Get's the value of a hashkey

  matchline=grep("\A".key."\A",hash,"tb");

  value=ntharg(matchline,"2","\B","\n");

  value=replace(value,"\N","\n");

  return(value);

}

function hash_delete(hash,key)

{

  # Get's the rest from the hash

  return(grep("\A".key."\A",hash,"tv"));

}

myhash="";

myhash=hash_set(myhash,"foo","bar");

myhash=hash_set(myhash,"bar","Hello World");

print("DUMP myhash\n".hash_dump(myhash)."\nEND OF DUMP\n\n");

print(hash_get(myhash,hash_get(myhash,"foo")));

How do state change actions work?

Here are the basics about state change actions:

· State change actions can be defined on computer level and application (actually instance) level.

· State change actions are executed by the console and will only be executed when the computer/instance changes state.

· If you defined a state change on instance level, then if one parameter of your instance parameters goes into alarm (which changes the state of the instance to alarm), the alarm state change action will be fired off. If a bit later another parameter will go into alarm, this will not cause the state change to be retriggered, because the instance will not change its state again!

· If you define a state change to a computer class, only computer objects will trigger the action. Also, computer objects don't get their state from parameters, even when everything is set to propagate. They get their state directly from the parents of those parameter's instances (applications). Therefore, there's no 'direct' way for a computer class to know which parameter put it into alarm.

· State changes can be OS commands only.

· State change actions are very different from from recovery actions, which are associated with parameters and event notification actions which are associated with event. The agent executes both recovery actions and event notification actions.

· State change actions do not have an "auto repeat until..." possibility, but that doesn't mean this functionality cannot be developed by you. For example: on ALARM/WARN state change, invoke some kind of OS process such as a C program, perl script, OS script, etc. which would continuously run and periodically take the desired action. You would then use the OK state change to execute another OS command that would stop the ALARM state change action. 

· If you define state change actions on a global level, and additionally on a localized level (for a specific computer/instance) then both the global and the local state change action will fire off in parallel.

· In a state change action, following macro's can be used:

	APP_TYPE
	Application name (e.g., ORACLE)

	COMP_STATUS
	Computer’s state (e.g., VOID, OK, WARNING)

	COMP_TYPE
	Computer type (e.g., SOLARIS)

	CON_INFO
	Network connection information (e.g., host name or net address)

	CON_STATUS
	Network connection state (e.g., OK, Soft Problem)

	CON_TYPE
	network connection type (UDP or TCP)

	DISPLAY_NAME
	Host name where the console is displayed

	HOME
	Home of this parameter, task, or command

	HOSTNAME
	Host name

	INSTANCE_ID
	Instance id (e.g., ORACLE SID)

	LANGUAGE
	The language to use in sound

	PASSWORD
	User’s password

	SID
	Instance id (e.g., ORACLE SID)

	STATUS
	Status of computer or task

	TIME
	Current time of day

	USERNAME
	User’s login name

	VOLUME
	Sound volume

	WORST_CHILD
	Worst child (parameter or instance) running on this object


These macros will be substituted prior to executing the OS Command.

Example: echo %{HOSTNAME}

How can I write a message dispatcher in PSL?

Sometimes you would like to dispatch messages that are retrieved by a single collector to multiple consumers.

There are multiple ways to achieve this, but to give you a head start; I added some code below that can help you in writing the method that you like most.

If you are looking for performance, it might be best to keep the messages in memory.

For sake of example I wrote the code based a collector of SNMP traps.  That means one of the parameters is collecting the traps (trap-receiver)... other parameters are reusing that info (trap-consumers). This is what your code might do on the side of the trap-receiver. 

# Trap receiving code

# replace this with your km name

kmname="/SNMP_TRAP";

# Replace this with the maxnumber of traps

maxtraps=999;

# Save maxtraps for the trap-consumers

set(kmname."/maxtraps",999);

# Find out what the last trap was.  Actually we will get

# The value of what the next_trap number should be

# This makes everything a lot easier to handle agent

# startup situations.

next_trap=get(kmname."/next_trap");

while (trapinfo=snmp_trap_receive())

{

  # Set the namespace variable

  set(kmname."/trapinfo_".next_trap,trapinfo);

  # Increase the next_trap number

  next_trap++;

  # And wrap around when maxtraps is reached

  next_trap%=maxtraps;

  # Let everyone know there is a new trap

  set(kmname."/next_trap",next_trap);

}

Your trap consumers (we'll have to give it a name), can then all do the following 

# Trap consumer code

# KM that contains the traps

trapkm="/SNMP_TRAP";

# Get the current status

#    How many traps have been received

next_trap=get(trapkm."/next_trap");

#    How many traps I have processed

my_next_trap=get("my_next_trap");

# Were there new traps ?

if (my_next_trap != next_trap)

{

  #    Get number of traps that are cached

  maxtraps=get(trapkm."/maxtraps");

  # If so, process them

  do

  {

    # Get the info of the trap.

    trapinfo=get(trapkm."/trapinfo_".my_next_trap);

    # Process the trap here

    # This is the user defined function for the consumer

    processtrap(trapinfo);

    # Increase the my_next_trap number

    my_next_trap++;

    # And wrap around when maxtraps is reached

    my_next_trap%=maxtraps;

  # Process till we caught up

  } until (my_next_trap == next_trap) 

  # Remember which traps we have processed

  set("my_next_trap",my_next_trap);

}

This code will process traps very fast. The collector will just receive traps and store them in memory; the trap consumers will process the received traps each time they are executed. If there are no new traps, the trap consumers will terminate quickly ... If there are new traps, they will be taken from memory.

You can change the trap "cache" in a very flexible way as well. It might be useful to check that you are not going to overwrite unprocessed traps. You can do that by having the "readers" set a flag with the last trap they read. The writer should not go above this boundary, or at least issue a warning if that happens.

I didn't add this to the code, since it would make it bigger and less readable, but it's reasonably easy to do.

Also right now the trap consumers have to be scheduled.  You can wrap their scheduling in a big while(1) loop and then use cond_* PSL functions to wake the consumers up.

Be careful about memory consumption... Since messages are cached in memory, you have to be careful about the size of your cache. 

Because the cache is circular, memory consumption will continue to grow until the cache is completely used. You should at least test the consumption at the time when the cache is completely filled and wraps around. You can use the deployment KM to do that

Does PATROL support SNMP v2?

Actually not, since the SNMP implementation in PATROL is only for v1.  Since the SNMPv2 protocol is mostly backward compatible with v1 the PATROL agent is able to talk to v2 agent.  

However, SNMP V2 introduces some new data types. A KM written to use these datatypes would not work.

 Also when you receive traps from an SNMPv2 agent, it's important to know that the PATROL agent will not understand “SNMPv2 Specific datatypes”.

New datatypes introduced in SNMPv2:

- Unsigned32

- Counter64

If you hit one of these datatypes, you will probably not see the result, because the agent can't decode the message properly.

Besides that SNMPv2 introduced a "bulk" get operation.  The agent will not respond to such requests.1

Libraries

How do I build a library?

You can build a library by using the standalone PSL interpreter/compiler.  First you have to create a syntactically correct PSL script.  Each of the functions you want to use in the program that will use the library have to be exported.  

Then you call psl –l myfile.psl and a myfile.lib will be created.

For example

File : mylib.psl

# Export section

export function myfunc;

# Function declaration section

function myfunc(foo,bar)

{

  print(foo.” says “.bar);

}

C:\> psl –l mylib.psl

will generate a file mylib.lib.  The following code can be used to test your library

File : tester.psl

requires mylib;

myfunc(“John”,”hello”);

After you run this using “psl tester.psl”, you will get the following output:



John says hello

How portable is a library?

All lib files are portable. If a library is compiled on one platform it should not be compiled anywhere else.

You just have to make sure the code you put in the lib file will execute on every platform (if you want it to run on every platform).

Just think of a lib file as a java class file which contains bytecode. A lib file is actually compiled PSL code and it will contain Quad-code (which is the low level instruction language for the PSL virtual machine).

You can create an "OS abstraction layer" for all the OS calls you would like to execute. For example

if (get("/appType") == "NT")

{

  # Use dir for NT systems

  dircmd="dir";

}

else

{

  # Use ls for UNIX

  dircmd="ls";

}

execute("OS",dircmd);

Of course if you want to support other platforms like VMS you will have to extend that list.

How does PATROL recognize the version of my library?

The version of a PSL binary is derived off the PSL Serial Number....

# PSL Serial No.: 962975710

This number is the same as the number a PATROL console puts at the top of a PSL script when it saves. This serial number ends up in the .lib and .bin file when compiled.

This library version is specifically used during commit when an ExtraFilesList has been specified.

During the processing of a LIB commit directive in the ExtraFilesList the console opens the library, obtains the serial number and sends it to the agent as its part of the commit handshake. The agent when it receives the version number opens the file on disk to compare the serial number of the library on disk with the serial number, which the console sent. If the numbers are the same the file is not transferred, otherwise they are.

If you do not use the console to edit, the PSL the serial number does not find it's way into the PSL script or the library.  In that case put a random serial number in the script. 

There are a couple of things to keep in mind when committing PSL libraries. The commit and the comparison only manipulate the libraries on DISK. If the agent has the library loaded that library is completely unaffected by any commit activity. The agent needs to be bounced in order for the committed library to be loaded and used. 

How can I check if I library was installed on the agent?

When the PSL compiler parses your PSL code and it finds a “requires” statement, the agent will try to load the library.

The fact that libraries are loaded during compilation causes some problems, because whenever a developer console connects to the agent, the PSL code will be send to the agent, but the libraries won’t (resulting in compiler errors).

Therefore it would be good to check if a library exists first, before launching a PSL script that requires it (could be done from prediscovery for example).

Before I show you code that will work, I will first show you a piece of code that will not work at all:

phome=get(“/patrolHome”);

libfile=phome.”/lib/mylib.lib”;

if (file(libfile))

{
  requires mylib;

}

else

{

  print(“Library not available, exiting …”);

  exit;

}

As I’ve mentioned before, it is the compiler that will see this requires statement and will load it.  If the library was not installed on the system, you will never get the error message, because the PSL script won’t even be compiled.

A piece of code that will do the trick is:

warn="";

alarm="";

PslExecute("LIBTEST","requires \"mylib1.lib\";\n".

                     "requires \"mylib2.lib\";",warn,alarm);

if (warn || alarm)

{

   print("KM not properly installed, libraries missing !");

   exit;

}

(D)COM & PATROL

How do I call the PslExecute method from COM?

Some people have noticed a problem when trying to execute PSL from VB with PSLExecute. The PSL code often include characters like "". VB uses these characters in such a manner that the execution of the PSL code fails.

This problem can be solved easily by replacing each single quote to a double quote.

For example: 

' Set Command

cmd = "set(""/tmpforProfCPU"",ntharg(ProfGetTotalCpu(),

           ""-"","" \t"","" ""));"

' And execute

vntRetValue = PatrolAgent.PslExecute("ProfCPUGet", cmd)

